Contents

Chapter 1

Library B_Unification.intro

1.1 Introduction

1.2 Unification

Before defining what unification is, there is some terminology to understand. A *term* is either a variable or a function applied to terms. By this definition, a constant term is just a nullary function. A *variable* is a symbol capable of taking on the value of any term. An examples of a term is $f(a, x)$, where f is a function of two arguments, a is a constant, and x is a variable. A term is *ground* if no variables occur in it. The last example is not a ground term but $f(a, a)$ would be.

A *substitution* is a mapping from variables to terms. The *domain* of a substitution is the set of variables that do not get mapped to themselves. The *range* is the set of terms the are mapped to by the domain. It is common for substitutions to be referred to as mappings from terms to terms. A substitution s can be extended to this form by defining $s'(u)$ for two cases of u. If u is a variable, then $s'(u) = s(u)$. If u is a function $f(u1, ..., un)$, then $s'(u) = f(s'(u1), ..., s'(un))$.

Unification is the process of solving a set of equations between two terms. The set of equations is referred to as a unification problem. The process of solving one of these problems can be classified by the set of terms considered and the equality of any two terms. The latter property is what distinguishes two broad groups of algorithms, namely syntactic and semantic unification. If two terms are only considered equal if they are identical, then the unification is syntactic. If two terms are equal with respect to an equational theory, then the unification is semantic.

The goal of unification is to solve equations, which means to produce a substitution that unifies those equations. A substitution s *unifies* an equation u =? v if applying s to both sides makes them equal $s(u) = s(v)$. In this case, we call s a *solution* or *unifier*.

The goal of a unification algorithm is not just to produce a unifier but to produce one that is most general. A substitution is a *most general unifier* or *mgu* of a problem if it is more general than every other solution to the problem. A substitution s is more general than s' if there exists a third substitution $\mathtt{t}$ such that $s'(u) = \mathtt{t}(s(u))$ for any term u.

1.2.1 Syntatic Unification

This is the simpler version of unification. For two terms to be considered equal they must be identical. For example, the terms $x \times y$ and $y \times x$ are not syntactically equal, but would be equal modulo commutativity of multiplication. (more about solving these problems / why simpler...)

1.2.2 Semantic Unification

This kind of unification involves an equational theory. Given a set of identities E, we write that two terms u and v are equal with regards to E as $u =E\ v$. This means that identities of E can be applied to u as u' and v as v' in some way to make them syntactically equal, $u' = v'$. As an example, let C be the set $\{f(x, y) = f(y, x)\}$. This theory C axiomatizes the commutativity of the function f. It would then make sense to write $f(a, x) =C\ f(x, a)$. In general, for an arbitrary E, the problem of E-unification is undecidable.

1.2.3 Boolean Unification

In this paper, we focus on unfication modulo Boolean ring theory, also referred to as B-unification. The allowed terms in this theory are the constants 0 and 1 and binary functions $+$ and $\times$. The set of identities B is defined as the set $\{x + y = y + x, (x + y) + z = x + (y + z), x + x = 0, 0 + x = x, x \times (y + z) = (x \times y) + (x \times z), x \times y = y \times x, (x \times y) \times z = x \times (y \times z), x \times x = x, 0 \times x = 0, 1 \times x = x\}$. This set is equivalent to the theory of real numbers with the addition of $x + x = 0$ and $x \times x = x$.

Although a unification problem is a set of equations between two terms, we will now show informally that a B-unification problem can be viewed as a single equation $\mathtt{t} = 0$. Given a problem in its normal form $\{s1 = t1, ..., sn = t2\}$, we can transform it into $\{s1 + t1 = 0, ..., sn + tn = 0\}$ using a simple fact. The equation $s = \mathtt{t}$ is equivalent to $s + \mathtt{t} = 0$ since adding $\mathtt{t}$ to both sides of the equation turns the right hand side into $\mathtt{t} + \mathtt{t}$ which simplifies to 0. Then, given a problem $\{t1 = 0, ..., tn = 0\}$, we can transform it into $\{(t1 + 1) \times ... \times (tn + 1) = 1\}$. Unifying both of these sets is equivalent because if any t1, ..., tn is 1 the problem is not unifiable. Otherwise, if every t1, ..., tn can be made to equal 0, then both problems will be solved.

1.3 Formal Verification

Formal verification is the term used to describe the act of verifying (or disproving) the correctness of software and hardware systems or theories. Formal verification consists of a set of techinques that perform static analysis on the behavior of a system, or the correctness of a theory. It differs to dynamic analysis that uses simulation to evaluate the correctness of a system.

Formal verification is used because it does not have to evaluate every possible case or state to determine if a system or theory meets all the preset logical conditions and rerquirements. Moreover, as design and software systems sizes have increased (along with their simulation times), verification teams have been looking for alternative methods of proving or disproving the correctness of a system in order to reduce the required time to perform a correctness check or evaluation.

1.3.1 Proof Assistance

A proof assistant is a software tool that is used to formulate and prove or disprove theorems in computer science or mathematical logic. They are also be called interactive theorem provers and they may also involve some type of proof and text editor that the user can use to form and prove and define theorems, lemmas , functions , etc. They facilitate that process by allowing the user to search definitions, terms and even provide some kind of guidance during the formulation or proof of a theorem.

1.3.2 Verifying Systems

1.3.3 Verifying Theories

1.4 Importance

1.5 Development

There are many different approaches that one could take to go about formalizing a proof of Boolean Unification algorithms, each with their own challenges. For this development, we have opted to base our work largely off chapter 10, *Equational Unification*, in *Term Rewriting and All That* by Franz Baader and Tobias Nipkow. Specifically, section 10.4, titled *Boolean Unification*, details Boolean rings, data structures to represent them, and two algorithms to perform unification in Boolean rings.

We chose to implement two data structures for representing the terms of a Boolean unification problem, and two algorithms for performing unification. The two data structures chosen are an inductive Term type and lists of lists representing polynomial-form terms. The two algorithms are Lowenheim's formula and successive variable elimination.

1.5.1 Data Structures

The data structure used to represent a Boolean unification problem completely changes the shape of both the unification algorithm and the proof of correctness, and is therefore a very important decision. For this development, we have selected two different representations of

Boolean rings – first as a "Term" inductive type, and then as lists of lists representing terms in polynomial form.

The Term inductive type, used in the proof of Lowenheim's algorithm, is very simple and rather intuitive – a term in a Boolean ring is one of 5 things:

- The number 0
- The number 1
- A variable
- Two terms added together
- Two terms multiplied together

In our development, variables are represented as natural numbers.

After defining terms like this, it is necessary to define a new equality relation, referred to as term equivalence, for comparing terms. With the term equivalence relation defined, it is easy to define ten axioms enabling the ten identities that hold true over terms in Boolean rings.

The inductive representation of terms in a Boolean ring is defined in the file *terms.v*. Unification over these terms is defined in *term_unif.v*.

The second representation, used in the proof of successive variable elimination, uses lists of lists of variables to represent terms in polynomial form. A monomial is a list of distinct variables multiplied together. A polynomial, then, is a list of distinct monomials added together. Variables are represented the same way, as natural numbers. The terms 0 and 1 are represented as the empty polynomial and the polynomial containing only the empty monomial, respectively.

The interesting part of the polynomial representation is how the ten identities are implemented. Rather than writing axioms enabling these transformations, we chose to implement the addition and multiplication operations in such a way to ensure these rules hold true, as described in *Term Rewriting*.

Addition is performed by cancelling out all repeated occurrences of monomials in the result of appending the two lists together (ie, x+x=0). This is equivalent to the symmetric difference in set theory, keeping only the terms that are in either one list or the other (but not both). Multiplication is slightly more complicated. The product of two polynomials is the result of multiplying all combinations of monomials in the two polynomials and removing all repeated monomials. The product of two monomials is the result of keeping only one copy of each repeated variable after appending the two together.

By defining the functions like this, and maintaining that the lists are sorted with no duplicates, we ensure that all 10 rules hold over the standard coq equivalence function. This of course has its own benefits and drawbacks, but lent itself better to the nature of successive variable elimination.

The polynomial representation is defined in the file *poly.v*. Unification over these polynomials is defined in *poly_unif.v*.

1.5.2 Algorithms

For unification algorithms, we once again followed the work laid out in *Term Rewriting and All That* and implemented both Lowenheim's algorithm and successive variable elimination.

The first solution, Lowenheim's algorithm, is built on top of the term inductive type. Lowenheim's is based on the idea that the Lowenheim formula can take a ground unifier of a Boolean unification problem and turn it into a most general unifier. The algorithm then of course first requires finding a ground solution, accomplished through brute force, which is then passed through the formula to create a most general unifier. Lowenheim's algorithm is implemented in the file *lowenheim.v*, and the proof of correctness is in *lowenheim_proof.v*.

The second algorithm, successive variable elimination, is built on top of the list-of-list polynomial approach. Successive variable elimination is built on the idea that by factoring variables out of the equation one-by-one, we can eventually reach a ground unifier. This unifier can then be built up with the variables that were previously eliminated until a most general unifier for the original unification problem is achieved. Successive variable elimination and its proof of correctness are both in *sve.v*.

Chapter 2

Library B_Unification.terms

```
Require Import Bool.
Require Import Omega.
Require Import EqNat.
Require Import List.
Require Import Setoid.
Import ListNotations.
```

2.1 Introduction

In order for any proofs to be constructed in Coq, we need to formally define the logic and data across which said proofs will operate. Since the heart of our analysis is concerned with the unification of Boolean equations, it stands to reason that we should articulate precisely how algebra functions with respect to Boolean rings. To attain this, we shall formalize what an equation looks like, how it can be composed inductively, and also how substitutions behave when applied to equations.

2.2 Terms

2.2.1 Definitions

We shall now begin describing the rules of Boolean arithmetic as well as the nature of Boolean equations. For simplicity's sake, from now on we shall be referring to equations as terms.

```
Definition var := nat.
```

```
Definition var_eq_dec := Nat.eq_dec.
```

A term, as has already been previously described, is now inductively declared to hold either a constant value, a single variable, a sum of terms, or a product of terms.

```
Inductive term: Type :=
```

| *T0* : *term*
| *T1* : *term*
| *VAR* : *var* → *term*
| *SUM* : *term* → *term* → *term*
| *PRODUCT* : *term* → *term* → *term*.

For convenience's sake, we define some shorthanded notation for readability.

```
Implicit Types x y z : term.
Implicit Types n m : var.
Notation "x + y" := (SUM x y) (at level 50, left associativity).
Notation "x * y" := (PRODUCT x y) (at level 40, left associativity).
```

2.2.2 Axioms

Now that we have informed Coq on the nature of what a term is, it is now time to propose a set of axioms that will articulate exactly how algebra behaves across Boolean rings. This is a requirement since the very act of unifying an equation is intimately related to solving it algebraically. Each of the axioms proposed below describe the rules of Boolean algebra precisely and in an unambiguous manner. None of these should come as a surprise to the reader; however, if one is not familiar with this form of logic, the rules regarding the summation and multiplication of identical terms might pose as a source of confusion.

For reasons of keeping Coq's internal logic consistent, we roll our own custom equivalence relation as opposed to simply using '='. This will provide a surefire way to avoid any odd errors from later cropping up in our proofs. Of course, by doing this we introduce some implications that we will need to address later.

```
Parameter eqv : term → term → Prop.
Infix " == " := eqv (at level 70).

Axiom sum_comm : ∀ x y, x + y == y + x.

Axiom sum_assoc : ∀ x y z, (x + y) + z == x + (y + z).

Axiom sum_id : ∀ x, T0 + x == x.

Axiom sum_x_x : ∀ x, x + x == T0.

Axiom mul_comm : ∀ x y, x × y == y × x.

Axiom mul_assoc : ∀ x y z, (x × y) × z == x × (y × z).

Axiom mul_x_x : ∀ x, x × x == x.

Axiom mul_T0_x : ∀ x, T0 × x == T0.

Axiom mul_id : ∀ x, T1 × x == x.

Axiom distr : ∀ x y z, x × (y + z) == (x × y) + (x × z).

Axiom term_sum_symmetric :
  ∀ x y z, x == y ↔ x + z == y + z.
```

```
Axiom term_product_symmetric :
  ∀ x y z, x == y ↔ x × z == y × z.

Axiom refl_comm :
∀ t1 t2, t1 == t2 → t2 == t1.

Hint Resolve sum_comm sum_assoc sum_x_x sum_id distr
             mul_comm mul_assoc mul_x_x mul_T0_x mul_id.
```

Now that the core axioms have been taken care of, we need to handle the implications posed by our custom equivalence relation. Below we inform Coq of the behavior of our equivalence relation with respect to rewrites during proofs.

```
Axiom eqv_ref : Reflexive eqv.
Axiom eqv_sym : Symmetric eqv.
Axiom eqv_trans : Transitive eqv.

Add Parametric Relation : term eqv
  reflexivity proved by @eqv_ref
  symmetry proved by @eqv_sym
  transitivity proved by @eqv_trans
  as eq_set_rel.

Axiom SUM_compat :
  ∀ x x', x == x' →
  ∀ y y', y == y' →
    (x + y) == (x' + y').

Axiom PRODUCT_compat :
  ∀ x x', x == x' →
  ∀ y y', y == y' →
    (x × y) == (x' × y').

Add Parametric Morphism : SUM with
  signature eqv ==> eqv ==> eqv as SUM_mor.
Proof.
exact SUM_compat.
Qed.

Add Parametric Morphism : PRODUCT with
  signature eqv ==> eqv ==> eqv as PRODUCT_mor.
Proof.
exact PRODUCT_compat.
Qed.

Hint Resolve eqv_ref eqv_sym eqv_trans SUM_compat PRODUCT_compat.
```

2.2.3 Lemmas

Since Coq now understands the basics of Boolean algebra, it serves as a good exercise for us to generate some further rules using Coq's proving systems. By doing this, not only do we gain some additional tools that will become handy later down the road, but we also test whether our axioms are behaving as we would like them to.

```
Lemma mul_x_x_plus_T1 :
  ∀ x, x × (x + T1) == T0.
Proof.
intros. rewrite distr. rewrite mul_x_x. rewrite mul_comm.
rewrite mul_id. apply sum_x_x.
Qed.
```

```
Lemma x_equal_y_x_plus_y :
  ∀ x y, x == y ↔ x + y == T0.
Proof.
intros. split.
- intros. rewrite H. rewrite sum_x_x. reflexivity.
- intros. rewrite term_sum_symmetric with (y := y) (z := y). rewrite sum_x_x.
  apply H.
Qed.
```

```
Hint Resolve mul_x_x_plus_T1.
Hint Resolve x_equal_y_x_plus_y.
```

These lemmas just serve to make certain rewrites regarding the core axioms less tedious to write. While one could certainly argue that they should be formulated as axioms and not lemmas due to their triviality, being pedantic is a good exercise.

```
Lemma sum_id_sym :
  ∀ x, x + T0 == x.
Proof.
intros. rewrite sum_comm. apply sum_id.
Qed.
```

```
Lemma mul_id_sym :
  ∀ x, x × T1 == x.
Proof.
intros. rewrite mul_comm. apply mul_id.
Qed.
```

```
Lemma mul_T0_x_sym :
  ∀ x, x × T0 == T0.
Proof.
intros. rewrite mul_comm. apply mul_T0_x.
Qed.
```

```
Lemma sum_assoc_opp :
```

```
  ∀ x y z, x + (y + z) == (x + y) + z.
Proof.
  intros. rewrite sum_assoc. reflexivity.
Qed.

Lemma mul_assoc_opp :
  ∀ x y z, x × (y × z) == (x × y) × z.
Proof.
  intros. rewrite mul_assoc. reflexivity.
Qed.

Lemma distr_opp :
  ∀ x y z, x × y + x × z == x × ( y + z).
Proof.
  intros. rewrite distr. reflexivity.
Qed.
```

2.3 Variable Sets

Now that the underlying behavior concerning Boolean algebra has been properly articulated to Coq, it is now time to begin formalizing the logic surrounding our meta reasoning of Boolean equations and systems. While there are certainly several approaches to begin this process, we thought it best to ease into things through formalizing the notion of a set of variables present in an equation.

2.3.1 Definitions

We now define a variable set to be precisely a list of variables; additionally, we include several functions for including and excluding variables from these variable sets. Furthermore, since uniqueness is not a property guaranteed by Coq lists and it has the potential to be desirable, we define a function that consumes a variable set and removes duplicate entries from it. For convenience, we also provide several examples to demonstrate the functionalities of these new definitions.

```
Definition var_set := list var.
Implicit Type vars: var_set.

Fixpoint var_set_includes_var (v : var) (vars : var_set) : bool :=
  match vars with
    | nil ⇒ false
    | n :: n' ⇒ if (beq_nat v n) then true else var_set_includes_var v n'
  end.

Fixpoint var_set_remove_var (v : var) (vars : var_set) : var_set :=
  match vars with
    | nil ⇒ nil
```

```
    | n :: n' ⇒ if (beq_nat v n) then (var_set_remove_var v n') else n :: (var_set_remove_var
v n')
  end.

Fixpoint var_set_create_unique (vars : var_set): var_set :=
  match vars with
    | nil ⇒ nil
    | n :: n' ⇒
    if (var_set_includes_var n n') then var_set_create_unique n'
    else n :: var_set_create_unique n'
  end.

Fixpoint var_set_is_unique (vars : var_set): bool :=
  match vars with
    | nil ⇒ true
    | n :: n' ⇒
    if (var_set_includes_var n n') then false
    else var_set_is_unique n'
  end.

Fixpoint term_vars (t : term) : var_set :=
  match t with
    | T0 ⇒ nil
    | T1 ⇒ nil
    | VAR x ⇒ x :: nil
    | PRODUCT x y ⇒ (term_vars x) ++ (term_vars y)
    | SUM x y ⇒ (term_vars x) ++ (term_vars y)
  end.

Definition term_unique_vars (t : term) : var_set :=
  (var_set_create_unique (term_vars t)).

Lemma vs_includes_true : ∀ (x : var) (lvar : list var),
  var_set_includes_var x lvar = true → In x lvar.
 Proof.
 intros.
  induction lvar.
  - simpl; intros.
  discriminate.
  - simpl in H. remember (beq_nat x a) as H2. destruct H2.
  + simpl. left. symmetry in HeqH2. pose proof beq_nat_true as H7. specialize (H7
x a HeqH2).
    symmetry in H7. apply H7.
  + specialize (IHlvar H). simpl. right. apply IHlvar.
Qed.

Lemma vs_includes_false : ∀ (x : var) (lvar : list var),
```

```
  var_set_includes_var x lvar = false → ¬ In x lvar.
 Proof.
 intros.
  induction lvar.
  - simpl; intros. unfold not. intros. destruct H0.
  - simpl in H. remember (beq_nat x a) as H2. destruct H2. inversion H.
    specialize (IHlvar H). firstorder. intuition. apply IHlvar. simpl in H0.
    destruct H0.
    { inversion HeqH2. symmetry in H2. pose proof beq_nat_false as H7. specialize
(H7 x a H2).
      rewrite H0 in H7. destruct H7. intuition. }
    { apply H0. }
Qed.

Lemma in_dup_and_non_dup :
 ∀ (x: var) (lvar : list var),
 In x lvar ↔ In x (var_set_create_unique lvar).
Proof.
 intros. split.
 - induction lvar.
  + intros. simpl in H. destruct H.
  + intros. simpl. remember(var_set_includes_var a lvar) as C. destruct C.
   { symmetry in HeqC. pose proof vs_includes_true as H7. specialize (H7 a lvar HeqC).
     simpl in H. destruct H.
    { rewrite H in H7. specialize (IHlvar H7). apply IHlvar. }
    { specialize (IHlvar H). apply IHlvar. }
   }
   { symmetry in HeqC. pose proof vs_includes_false as H7. specialize (H7 a lvar HeqC).
     simpl in H. destruct H.
    { simpl. left. apply H. }
    { specialize (IHlvar H). simpl. right. apply IHlvar. }
   }
 - induction lvar.
   + intros. simpl in H. destruct H.
   + intros. simpl in H. remember(var_set_includes_var a lvar) as C. destruct C.
     { symmetry in HeqC. pose proof vs_includes_true as H7. specialize (H7 a lvar
HeqC).
      specialize (IHlvar H). simpl. right. apply IHlvar. }
     { symmetry in HeqC. pose proof vs_includes_false as H7. specialize (H7 a lvar
HeqC).
       simpl in H. destruct H.
      { simpl. left. apply H. }
      { specialize (IHlvar H). simpl. right. apply IHlvar. } }
```

Qed.

2.3.2 Examples

```
Example var_set_create_unique_ex1 :
  var_set_create_unique [0;5;2;1;1;2;2;9;5;3] = [0;1;2;9;5;3].
Proof.
simpl. reflexivity.
Qed.
```

```
Example var_set_is_unique_ex1 :
  var_set_is_unique [0;2;2;2] = false.
Proof.
simpl. reflexivity.
Qed.
```

```
Example term_vars_ex1 :
  term_vars (VAR 0 + VAR 0 + VAR 1) = [0;0;1].
Proof.
simpl. reflexivity.
Qed.
```

```
Example term_vars_ex2 :
  In 0 (term_vars (VAR 0 + VAR 0 + VAR 1)).
Proof.
simpl. left. reflexivity.
Qed.
```

2.4 Ground Terms

Seeing as we just outlined the definition of a variable set, it seems fair to now formalize the definition of a ground term, or in other words, a term that has no variables and whose variable set is the empty set.

2.4.1 Definitions

A ground term is a recursively defined proposition that is only True if and only if no variable appears in it; otherwise it will be a False proposition and no longer a ground term.

```
Fixpoint ground_term (t : term) : Prop :=
  match t with
    | VAR x ⇒ False
    | SUM x y ⇒ (ground_term x) ∧ (ground_term y)
    | PRODUCT x y ⇒ (ground_term x) ∧ (ground_term y)
```

```
  | _ ⇒ True
end.
```

2.4.2 Lemmas

Our first real lemma (shown below), articulates an important property of ground terms: all ground terms are equvialent to either 0 or 1. This curious property is a direct result of the fact that these terms possess no variables and additioanlly because of the axioms of Boolean algebra.

```
Lemma ground_term_equiv_T0_T1 :
  ∀ x, (ground_term x) → (x == T0 ∨ x == T1).
Proof.
intros. induction x.
- left. reflexivity.
- right. reflexivity.
- contradiction.
- inversion H. destruct IHx1; destruct IHx2; auto. rewrite H2. left. rewrite sum_id.
apply H3.
rewrite H2. rewrite H3. rewrite sum_id. right. reflexivity.
rewrite H2. rewrite H3. right. rewrite sum_comm. rewrite sum_id. reflexivity.
rewrite H2. rewrite H3. rewrite sum_x_x. left. reflexivity.
- inversion H. destruct IHx1; destruct IHx2; auto. rewrite H2. left. rewrite
mul_T0_x. reflexivity.
rewrite H2. left. rewrite mul_T0_x. reflexivity.
rewrite H3. left. rewrite mul_comm. rewrite mul_T0_x. reflexivity.
rewrite H2. rewrite H3. right. rewrite mul_id. reflexivity.
Qed.
```

This lemma, while intuitively obvious by definition, nonetheless provides a formal bridge between the world of ground terms and the world of variable sets.

```
Lemma ground_term_has_empty_var_set :
  ∀ x, (ground_term x) → (term_vars x) = [].
Proof.
intros. induction x.
- simpl. reflexivity.
- simpl. reflexivity.
- contradiction.
- firstorder. unfold term_vars. unfold term_vars in H2. rewrite H2. unfold term_vars
in H1. rewrite H1. simpl. reflexivity.
- firstorder. unfold term_vars. unfold term_vars in H2. rewrite H2. unfold term_vars
in H1. rewrite H1. simpl. reflexivity.
Qed.
```

2.4.3 Examples

Here are some examples to show that our ground term definition is working appropriately.

```
Example ex_gt1 :
  (ground_term (T0 + T1)).
Proof.
simpl. split.
- reflexivity.
- reflexivity.
Qed.
```

```
Example ex_gt2 :
  (ground_term (VAR 0 × T1)) → False.
Proof.
simpl. intros. destruct H. apply H.
Qed.
```

2.5 Substitutions

It is at this point in our Coq development that we begin to officially define the principal action around which the entirety of our efforts are centered: the act of substituting variables with other terms. While substitutions alone are not of great interest, their emergent properties as in the case of whether or not a given substitution unifies an equation are of substantial importance to our later research.

2.5.1 Definitions

Here we define a substitution to be a list of ordered pairs where each pair represents a variable being mapped to a term. For sake of clarity these ordered pairs shall be referred to as replacements from now on and as a result, substitutions should really be considered to be lists of replacements.

```
Definition replacement := (prod var term).
```

```
Definition subst := list replacement.
```

```
Implicit Type s : subst.
```

Our first function, find_replacement, is an auxilliary to apply_subst. This function will search through a substitution for a specific variable, and if found, returns the variable's associated term.

```
Fixpoint find_replacement (x : var) (s : subst) : term :=
  match s with
  | nil ⇒ VAR x
  | r :: r' ⇒
```

```
    if beq_nat (fst r) x then (snd r)
    else
      (find_replacement x r')
  end.
```

The apply_subst function will take a term and a substitution and will produce a new term reflecting the changes made to the original one.

```
Fixpoint apply_subst (t : term) (s : subst) : term :=
  match t with
  | T0 ⇒ T0
  | T1 ⇒ T1
  | VAR x ⇒ (find_replacement x s)
  | PRODUCT x y ⇒ PRODUCT (apply_subst x s) (apply_subst y s)
  | SUM x y ⇒ SUM (apply_subst x s) (apply_subst y s)
  end.
```

For reasons of completeness, it is useful to be able to generate identity substitutions; namely, substitutions that map the variables of a term's variable set to themselves.

```
Fixpoint build_id_subst (lvar : var_set) : subst :=
  match lvar with
  | nil ⇒ nil
  | v :: v' ⇒ (cons (v , (VAR v))
                          (build_id_subst v'))
  end.
```

Since we now have the ability to generate identity substitutions, we should now formalize a general proposition for testing whether or not a given substitution is an identity substitution of a given term.

```
Definition subst_equiv (s1 s2: subst) : Prop :=
  ∀ r, In r s1 ↔ In r s2.

Definition subst_is_id_subst (t : term) (s : subst) : Prop :=
  (subst_equiv (build_id_subst (term_vars t)) s).
```

2.5.2 Lemmas

Having now outlined the functionality of a subsitution, let us now begin to analyze some implications of its form and composition by proving some lemmas.

```
Lemma apply_subst_compat : ∀ (t t' : term),
      t == t' → ∀ (sigma: subst), (apply_subst t sigma) == (apply_subst t' sigma).
Proof.
Admitted.

Add Parametric Morphism : apply_subst with
      signature eqv ==> eq ==> eqv as apply_subst_mor.
```

```
Proof.
  exact apply_subst_compat.
Qed.
```

An easy thing to prove right off the bat is that ground terms, i.e. terms with no variables, cannot be modified by applying substitutions to them. This will later prove to be very relevant when we begin to talk about unification.

```
Lemma ground_term_cannot_subst :
  ∀ x, (ground_term x) → (∀ s, apply_subst x s == x).
Proof.
intros. induction s.
  - apply ground_term_equiv_T0_T1 in H. destruct H.
  + rewrite H. simpl. reflexivity.
  + rewrite H. simpl. reflexivity.
  - apply ground_term_equiv_T0_T1 in H. destruct H. rewrite H.
    + simpl. reflexivity.
    + rewrite H. simpl. reflexivity.
Qed.
```

A fundamental property of substitutions is their distributivity and associativity across the summation and multiplication of terms. Again the importance of these proofs will not become apparent until we talk about unification.

```
Lemma subst_distribution :
  ∀ s x y, apply_subst x s + apply_subst y s == apply_subst (x + y) s.
Proof.
intro. induction s. simpl. intros. reflexivity. intros. simpl. reflexivity.
Qed.

Lemma subst_associative :
  ∀ s x y, apply_subst x s × apply_subst y s == apply_subst (x × y) s.
Proof.
intro. induction s. intros. reflexivity. intros. simpl. reflexivity.
Qed.

Lemma subst_sum_distr_opp :
  ∀ s x y, apply_subst (x + y) s == apply_subst x s + apply_subst y s.
Proof.
  intros.
  apply refl_comm.
  apply subst_distribution.
Qed.

Lemma subst_mul_distr_opp :
  ∀ s x y, apply_subst (x × y) s == apply_subst x s × apply_subst y s.
Proof.
  intros.
```

```
  apply refl_comm.
  apply subst_associative.
Qed.

Lemma var_subst:
  ∀ (v : var) (ts : term) ,
  (apply_subst (VAR v) (cons (v , ts) nil) ) == ts.
Proof.
intros. simpl. destruct (beq_nat v v) eqn: e. apply beq_nat_true in e.
reflexivity. apply beq_nat_false in e. firstorder.
Qed.
```

Given that we have a definition for identity substitutions, we should prove that identity substitutions do not modify a term.

```
Lemma id_subst:
  ∀ (t : term) (l : var_set),
  apply_subst t (build_id_subst l) == t.
Proof.
intros. induction t.
{
  simpl. reflexivity.
}
{
  simpl. reflexivity.
}
{
  simpl. induction l.
  {
    simpl. reflexivity.
  }
  {
    simpl. destruct (beq_nat a v) eqn: e.
    {
      apply beq_nat_true in e. rewrite e. reflexivity.
    }
    {
      apply IHl.
    }
  }
}
{
  simpl. rewrite IHt1. rewrite IHt2. reflexivity.
}
{
```

```
  simpl. rewrite IHt1. rewrite IHt2. reflexivity.
}
Qed.
```

2.5.3 Examples

Here are some examples showcasing the nature of applying substitutions to terms.

```
Example subst_ex1 :
  (apply_subst (T0 + T1) []) == T0 + T1.
Proof.
intros. reflexivity.
Qed.
```

```
Example subst_ex2 :
  (apply_subst (VAR 0 × VAR 1) [(0, T0)]) == T0.
Proof.
intros. simpl. apply mul_T0_x.
Qed.
```

2.6 Unification

Now that we have established the concept of term substitutions in Coq, it is time for us to formally define the concept of Boolean unification. Unification, in its most literal sense, refers to the act of applying a substitution to terms in order to make them equivalent to each other. In other words, to say that two terms are unifiable is to really say that there exists a substitution such that the two terms are equal. Interestingly enough, we can abstract this concept further to simply saying that a single term is unifiable if there exists a substitution such that the term will be equivalent to 0. By doing this abstraction, we can prove that equation solving and unification are essentially the same fundamental problem.

Below is the initial definition for unification, namely that two terms can be unified to be equivalent to one another. By starting here we will show each step towards abstracting unification to refer to a single term.

```
Definition unifies (a b : term) (s : subst) : Prop :=
  (apply_subst a s) == (apply_subst b s).
```

Here is a simple example demonstrating the concept of testing whether two terms are unified by a substitution.

```
Example ex_unif1 :
  unifies (VAR 0) (VAR 1) ((0, T1) :: (1, T1) :: nil).
Proof.
unfold unifies. simpl. reflexivity.
Qed.
```

Now we are going to show that moving both terms to one side of the equivalence relation through addition does not change the concept of unification.

```
Definition unifies_T0 (a b : term) (s : subst) : Prop :=
  (apply_subst a s) + (apply_subst b s) == T0.
```

```
Lemma unifies_T0_equiv :
  ∀ x y s, unifies x y s ↔ unifies_T0 x y s.
Proof.
intros. split.
{
  intros. unfold unifies_T0. unfold unifies in H. rewrite H.
  rewrite sum_x_x. reflexivity.
}
{
  intros. unfold unifies_T0 in H. unfold unifies.
  rewrite term_sum_symmetric with (x := apply_subst x s + apply_subst y s)
  (z := apply_subst y s) in H. rewrite sum_id in H.
  rewrite sum_comm in H.
  rewrite sum_comm with (y := apply_subst y s) in H.
  rewrite ← sum_assoc in H.
  rewrite sum_x_x in H.
  rewrite sum_id in H.
  apply H.
}
Qed.
```

Now we can define what it means for a substitution to be a unifier for a given term.

```
Definition unifier (t : term) (s : subst) : Prop :=
  (apply_subst t s) == T0.
```

```
Example unifier_ex1 :
  (unifier (VAR 0) ((0, T0) :: nil)).
Proof.
unfold unifier. simpl. reflexivity.
Qed.
```

To ensure our efforts were not in vain, let us now prove that this last abstraction of the unification problem is still equivalent to the original.

```
Lemma unifier_distribution :
  ∀ x y s, (unifies_T0 x y s) ↔ (unifier (x + y) s).
Proof.
intros. split.
{
  intros. unfold unifies_T0 in H. unfold unifier.
  rewrite ← H. symmetry. apply subst_distribution.
```

```
}
{
  intros. unfold unifies_T0. unfold unifier in H.
  rewrite ← H. apply subst_distribution.
}
Qed.
```

```
Lemma unifier_subset_imply_superset :
  ∀ s t r, unifier t s → unifier t (r :: s).
Proof.
intros. induction t.
{
  unfold unifier in *. simpl. reflexivity.
}
{
  unfold unifier in *. simpl in *. apply H.
}
{
  unfold unifier in *. simpl in *. destruct beq_nat.
Admitted.
```

Lastly let us define a term to be unifiable if there exists a substitution that unifies it.

```
Definition unifiable (t : term) : Prop :=
  ∃ s, unifier t s.
```

```
Example unifiable_ex1 :
  ∃ x, unifiable (x + T1).
Proof.
∃ (T1). unfold unifiable. unfold unifier.
∃ nil. simpl. rewrite sum_x_x. reflexivity.
Qed.
```

2.7 Most General Unifier

```
Definition substitution_composition (s s' delta : subst) (t : term) : Prop :=
  ∀ (x : var), apply_subst (apply_subst (VAR x) s) delta == apply_subst (VAR x) s' .
```

```
Definition more_general_substitution (s s': subst) (t : term) : Prop :=
  ∃ delta, substitution_composition s s' delta t.
```

```
Definition most_general_unifier (t : term) (s : subst) : Prop :=
  (unifier t s) → (∀ (s' : subst), unifier t s' → more_general_substitution s s' t ).
```

```
Definition reproductive_unifier (t : term) (sig : subst) : Prop :=
  unifier t sig →
```

```
  ∀ (tau : subst) (x : var),
  unifier t tau →
  (apply_subst (apply_subst (VAR x) sig ) tau) == (apply_subst (VAR x) tau).
Lemma reproductive_is_mgu : ∀ (t : term) (u : subst),
  reproductive_unifier t u →
  most_general_unifier t u.
Proof.
 intros. unfold most_general_unifier. unfold reproductive_unifier in H.
  unfold more_general_substitution . unfold substitution_composition.
  intros. specialize (H H0). ∃ s' . intros. specialize (H s' x). specialize (H
H1). apply H.
Qed.
```

2.8 Auxilliary Computational Operations and Simplifications

These functions below will come in handy later during the Lowenheim formula proof.

```
Fixpoint identical (a b: term) : bool :=
  match a , b with
    | T0, T0 ⇒ true
    | T0, _ ⇒ false
    | T1 , T1 ⇒ true
    | T1 , _ ⇒ false
    | VAR x , VAR y ⇒ if beq_nat x y then true else false
    | VAR x, _ ⇒ false
    | PRODUCT x y, PRODUCT x1 y1 ⇒ if ((identical x x1) && (identical y y1)) then
true
                                   else false
    | PRODUCT x y, _ ⇒ false
    | SUM x y, SUM x1 y1 ⇒ if ((identical x x1) && (identical y y1)) then true
                                   else false
    | SUM x y, _ ⇒ false
  end.
Definition plus_one_step (a b : term) : term :=
  match a, b with
    | T0, _ ⇒ b
    | T1, T0 ⇒ T1
    | T1, T1 ⇒ T0
    | T1 , _ ⇒ SUM a b
    | VAR x , T0 ⇒ a
```

```
    | VAR x , _ ⇒ if identical a b then T0 else SUM a b
    | PRODUCT x y , T0 ⇒ a
    | PRODUCT x y, _ ⇒ if identical a b then T0 else SUM a b
    | SUM x y , T0 ⇒ a
    | SUM x y, _ ⇒ if identical a b then T0 else SUM a b
  end.

Definition mult_one_step (a b : term) : term :=
  match a, b with
    | T0, _ ⇒ T0
    | T1 , _ ⇒ b
    | VAR x , T0 ⇒ T0
    | VAR x , T1 ⇒ a
    | VAR x , _ ⇒ if identical a b then a else PRODUCT a b
    | PRODUCT x y , T0 ⇒ T0
    | PRODUCT x y , T1 ⇒ a
    | PRODUCT x y, _ ⇒ if identical a b then a else PRODUCT a b
    | SUM x y , T0 ⇒ T0
    | SUM x y , T1 ⇒ a
    | SUM x y, _ ⇒ if identical a b then a else PRODUCT a b
  end.

Fixpoint simplify (t : term) : term :=
  match t with
    | T0 ⇒ T0
    | T1 ⇒ T1
    | VAR x ⇒ VAR x
    | PRODUCT x y ⇒ mult_one_step (simplify x) (simplify y)
    | SUM x y ⇒ plus_one_step (simplify x) (simplify y)
  end.

Fixpoint Simplify_N (t : term) (counter : nat): term :=
  match counter with
    | O ⇒ t
    | S n' ⇒ (Simplify_N (simplify t) n')
  end.
```

Chapter 3

Library B_Unification.lowenheim_formula

```
Require Export terms.

Require Import List.
Import ListNotations.

Fixpoint build_on_list_of_vars (list_var : var_set) (s : term) (sig1 : subst) (sig2 : subst) :
subst :=
  match list_var with
    | nil => nil
    | v' :: v =>
       (cons (v' , (s + T1) × (apply_subst (VAR v') sig1 ) + s × (apply_subst (VAR v' )
sig2 ) )
             (build_on_list_of_vars v s sig1 sig2) )
  end.

Definition build_lowenheim_subst (t : term) (tau : subst) : subst :=
  build_on_list_of_vars (term_unique_vars t) t (build_id_subst (term_unique_vars t)) tau.
```

2.2 Lowenheim's algorithm

```
Definition update_term (t : term) (s' : subst) : term :=
  (simplify (apply_subst t s' ) ).

Definition term_is_T0 (t : term) : bool :=
  (identical t T0).

Inductive subst_option: Type :=
    | Some_subst : subst -> subst_option
    | None_subst : subst_option.

Fixpoint rec_subst (t : term) (vars : var_set) (s : subst) : subst :=
```

```
  match vars with
    | nil ⇒ s
    | v' :: v ⇒
        if (term_is_T0
              (update_term (update_term t (cons (v' , T0) s) )
                           (rec_subst (update_term t (cons (v' , T0) s) )
                                      v (cons (v' , T0) s)) )
                           )
          then
              (rec_subst (update_term t (cons (v' , T0) s) )
                         v (cons (v' , T0) s))
       else
          if (term_is_T0
               (update_term (update_term t (cons (v' , T1) s) )
                            (rec_subst (update_term t (cons (v' , T1) s) )
                                       v (cons (v' , T1) s)) ) )
          then
              (rec_subst (update_term t (cons (v' , T1) s) )
                         v (cons (v' , T1) s))
          else
              (rec_subst (update_term t (cons (v' , T0) s) )
                         v (cons (v' , T0) s))
    end.

Compute (rec_subst ((VAR 0) × (VAR 1)) (cons 0 (cons 1 nil)) nil) .

Fixpoint find_unifier (t : term) : subst_option :=
  match (update_term t (rec_subst t (term_unique_vars t) nil) ) with
    | T0 ⇒ Some_subst (rec_subst t (term_unique_vars t) nil)
    | _ ⇒ None_subst
  end.

Compute (find_unifier ((VAR 0) × (VAR 1))).
Compute (find_unifier ((VAR 0) + (VAR 1))).
Compute (find_unifier ((VAR 0) + (VAR 1) + (VAR 2) + T1 + (VAR 3) × ( (VAR 2) + (VAR
0)) )).

Definition Lowenheim_Main (t : term) : subst_option :=
  match (find_unifier t) with
    | Some_subst s ⇒ Some_subst (build_lowenheim_subst t s)
    | None_subst ⇒ None_subst
  end.

Compute (find_unifier ((VAR 0) × (VAR 1)) ) .

Compute (Lowenheim_Main ((VAR 0) × (VAR 1))).
```

```
Compute (Lowenheim_Main ((VAR 0) + (VAR 1)) ).
Compute (Lowenheim_Main ((VAR 0) + (VAR 1) + (VAR 2) + T1 + (VAR 3) × ( (VAR 2) +
(VAR 0)) ) ).
Compute (Lowenheim_Main (T1)).
Compute (Lowenheim_Main (( VAR 0) + (VAR 0) + T1)).
```

2.3 Lowenheim testing

```
Definition Test_find_unifier (t : term) : bool :=
  match (find_unifier t) with
    | Some_subst s ⇒
      (term_is_T0 (update_term t s))
    | None_subst ⇒ true
  end.

Compute (Test_find_unifier (T1)).
Compute (Test_find_unifier ((VAR 0) × (VAR 1))).
Compute (Test_find_unifier ((VAR 0) + (VAR 1) + (VAR 2) + T1 + (VAR 3) × ( (VAR 2) +
(VAR 0)) )).
Definition apply_lowenheim_main (t : term) : term :=
  match (Lowenheim_Main t) with
  | Some_subst s ⇒ (apply_subst t s)
  | None_subst ⇒ T1
  end.
Compute (Lowenheim_Main ((VAR 0) × (VAR 1) )).
Compute (apply_lowenheim_main ((VAR 0) × (VAR 1) ) ).
Compute (Lowenheim_Main ((VAR 0) + (VAR 1) )).
Compute (apply_lowenheim_main ((VAR 0) + (VAR 1) ) ).
```

Chapter 4

Library B_Unification.lowenheim_proof

```
Require Export lowenheim_formula.

Require Export EqNat.
Require Import List.
Import ListNotations.
Import Coq.Init.Tactics.
Require Export Classical_Prop.

Require Export lowenheim_formula.
```

3.1 Declarations and their lemmas useful for the proof

```
Definition sub_term (t : term) (t' : term) : Prop :=
  ∀ (x : var ),
  (In x (term_unique_vars t) ) → (In x (term_unique_vars t')) .

Lemma sub_term_id :
  ∀ (t : term),
  sub_term t t.
Proof.
 intros. firstorder.
Qed.

Lemma term_vars_distr :
∀ (t1 t2 : term),
 (term_vars (t1 + t2)) = (term_vars t1) ++ (term_vars t2).
Proof.
 intros.
 induction t2.
 - simpl. reflexivity.
 - simpl. reflexivity.
```

```
- simpl. reflexivity.
- simpl. reflexivity.
- simpl. reflexivity.
Qed.

Lemma tv_h1:
∀ (t1 t2 : term) ,
∀ (x : var),
 (In x (term_vars t1)) → (In x (term_vars (t1 + t2))).
Proof.
intros. induction t2.
- simpl. rewrite app_nil_r. apply H.
- simpl. rewrite app_nil_r. apply H.
- simpl.  pose proof in_or_app as H1.  specialize (H1 var (term_vars t1) [v] x).
firstorder.
- rewrite term_vars_distr. apply in_or_app. left. apply H.
- rewrite term_vars_distr. apply in_or_app. left. apply H.
Qed.

Lemma tv_h2:
∀ (t1 t2 : term) ,
∀ (x : var),
 (In x (term_vars t2)) → (In x (term_vars (t1 + t2))).
Proof.
intros. induction t1.
- simpl. apply H.
- simpl. apply H.
- simpl. pose proof in_or_app as H1. right. apply H.
- rewrite term_vars_distr. apply in_or_app. right. apply H.
- rewrite term_vars_distr. apply in_or_app. right. apply H.
Qed.

Lemma helper_2a:
  ∀ (t1 t2 t' : term),
  sub_term (t1 + t2) t' → sub_term t1 t'.
Proof.
 intros. unfold sub_term in *. intros. specialize (H x).
 pose proof in_dup_and_non_dup as H10. unfold term_unique_vars. unfold term_unique_vars
in *.
 pose proof tv_h1 as H7.  apply H. specialize (H7 t1 t2 x).  specialize (H10 x
(term_vars (t1 + t2))). destruct H10 .
 apply H1.  apply H7.  pose proof in_dup_and_non_dup as H10.  specialize (H10 x
(term_vars t1)). destruct H10.
 apply H4. apply H0.
Qed.
```

```
Lemma helper_2b:
  ∀ (t1 t2 t' : term),
  sub_term (t1 + t2) t' → sub_term t2 t'.
Proof.
intros. unfold sub_term in *. intros. specialize (H x).
pose proof in_dup_and_non_dup as H10. unfold term_unique_vars. unfold term_unique_vars
in *.
 pose proof tv_h2 as H7. apply H. specialize (H7 t1 t2 x). specialize (H10 x
(term_vars (t1 + t2))). destruct H10 .
 apply H1. apply H7. pose proof in_dup_and_non_dup as H10. specialize (H10 x
(term_vars t2)). destruct H10.
 apply H4. apply H0.
Qed.

Lemma elt_in_list:
 ∀ (x: var) (a : var) (l : list var),
  (In x (a::l)) →
  x = a ∨ (In x l).
Proof.
intros.
pose proof in_inv as H1.
specialize (H1 var a x l H).
destruct H1.
 - left. symmetry in H0. apply H0.
 - right. apply H0.
Qed.

Lemma elt_not_in_list:
 ∀ (x: var) (a : var) (l : list var),
  ¬ (In x (a::l)) →
  x ≠ a ∧ ¬ (In x l).
Proof.
intros.
pose proof not_in_cons. specialize (H0 var x a l). destruct H0.
specialize (H0 H). apply H0.
Qed.

Lemma in_list_of_var_term_of_var:
∀ (x : var),
  In x (term_unique_vars (VAR x)).
Proof.
intros. simpl. left. intuition.
Qed.

Lemma var_in_out_list:
```

```
  ∀ (x : var) (lvar : list var),
  (In x lvar) ∨ ¬ (In x lvar).
Proof.
 intros.
 pose proof classic as H1. specialize (H1 (In x lvar)). apply H1.
Qed.
```

3.2 Proof that Lownheim's algorithm unifes a given term

```
Lemma helper1_easy:
 ∀ (x: var) (lvar : list var) (sig1 sig2 : subst) (s : term),
 (In x lvar) →
  apply_subst (VAR x) (build_on_list_of_vars lvar s sig1 sig2)
  ==
  apply_subst (VAR x) (build_on_list_of_vars (cons x nil) s sig1 sig2).
Proof.
 intros.
 induction lvar.
 - simpl. simpl in H. destruct H.
 - apply elt_in_list in H. destruct H.
  + simpl. destruct (beq_nat a x) as []eqn:?.
   { apply beq_nat_true in Heqb. destruct (beq_nat x x) as []eqn:?.
    { rewrite H. reflexivity. }
    { apply beq_nat_false in Heqb.
      { destruct Heqb. }
      { rewrite Heqb. apply Heqb0. } }}
   { simpl in IHlvar. apply IHlvar. symmetry in H. rewrite H in Heqb.
    apply beq_nat_false in Heqb. destruct Heqb. intuition. }
  + destruct (beq_nat a x) as []eqn:?.
    { apply beq_nat_true in Heqb. symmetry in Heqb. rewrite Heqb in IHlvar. rewrite
Heqb.
        simpl in IHlvar. simpl. destruct (beq_nat a a) as []eqn:?.
     { reflexivity. }
     { apply IHlvar. rewrite Heqb in H. apply H. }}
    { apply beq_nat_false in Heqb. simpl. destruct (beq_nat a x) as []eqn:?.
     { apply beq_nat_true in Heqb0. rewrite Heqb0 in Heqb. destruct Heqb. intuition.
}
     { simpl in IHlvar. apply IHlvar. apply H. }}
Qed.

Lemma helper_1:
∀ (t' s : term) (v : var) (sig1 sig2 : subst),
  sub_term (VAR v) t' →
  apply_subst (VAR v) (build_on_list_of_vars (term_unique_vars t') s sig1 sig2)
  ==
```

```
  apply_subst (VAR v) (build_on_list_of_vars (term_unique_vars (VAR v)) s sig1 sig2).
Proof.
 intros. unfold sub_term in H. specialize (H v). pose proof in_list_of_var_term_of_var
as H3.
 specialize (H3 v). specialize (H H3). pose proof helper1_easy as H2.
 specialize (H2 v (term_unique_vars t') sig1 sig2 s). apply H2. apply H.
Qed.

Lemma subs_distr_vars_ver2 :
  ∀ (t t' : term) (s : term) (sig1 sig2 : subst),
  (sub_term t t') →
  apply_subst t (build_on_list_of_vars (term_unique_vars t') s sig1 sig2)
    ==
  (s + T1) × (apply_subst t sig1) + s × (apply_subst t sig2).
Proof.
 intros. generalize dependent t'. induction t.
  - intros t'. repeat rewrite ground_term_cannot_subst.
     + rewrite mul_comm with (x := s + T1). rewrite distr. repeat rewrite mul_T0_x.
rewrite mul_comm with (x := s).
       rewrite mul_T0_x. repeat rewrite sum_x_x. reflexivity.
     + unfold ground_term. reflexivity.
     + unfold ground_term. reflexivity.
     + unfold ground_term. reflexivity.
  - intros t'. repeat rewrite ground_term_cannot_subst.
     + rewrite mul_comm with (x := s + T1). rewrite mul_id. rewrite mul_comm with
(x := s). rewrite mul_id. rewrite sum_comm with (x := s).
       repeat rewrite sum_assoc. rewrite sum_x_x. rewrite sum_comm with (x := T1).
rewrite sum_id. reflexivity.
     + unfold ground_term. reflexivity.
     + unfold ground_term. reflexivity.
     + unfold ground_term. reflexivity.
  - intros. rewrite helper_1 .
     + unfold term_unique_vars. unfold term_vars. unfold var_set_create_unique. unfold
var_set_includes_var. unfold build_on_list_of_vars.
     rewrite var_subst. reflexivity.
     + apply H.
  - intros. specialize (IHt1 t'). specialize (IHt2 t'). repeat rewrite subst_sum_distr_opp.
       rewrite IHt1. rewrite IHt2.
     + rewrite distr. rewrite distr. repeat rewrite sum_assoc. rewrite sum_comm with
(x := (s + T1) × apply_subst t2 sig1)
        (y := (s × apply_subst t1 sig2 + s × apply_subst t2 sig2)). repeat rewrite sum_assoc.
        rewrite sum_comm with (x := s × apply_subst t2 sig2 ) (y := (s + T1) × apply_subst
t2 sig1).
```

repeat rewrite *sum_assoc*. reflexivity.
+ pose helper_2b as $H2$. specialize ($H2$ $t1$ $t2$ t'). apply $H2$. apply H.
+ pose helper_2a as $H2$. specialize ($H2$ $t1$ $t2$ t'). apply $H2$. apply H.
- intros. specialize ($IHt1$ t'). specialize ($IHt2$ t'). repeat rewrite subst_mul_distr_opp. rewrite $IHt1$. rewrite $IHt2$.
+ rewrite *distr*. rewrite *mul_comm* with (y := ((s + T1) $\times$ apply_subst $t2$ $sig1$)).
rewrite *distr*. rewrite *mul_comm* with (y := (s $\times$ apply_subst $t2$ $sig2$)). rewrite *distr*.
repeat rewrite *mul_assoc*. repeat rewrite *mul_comm* with (x := apply_subst $t2$ $sig1$).
repeat rewrite *mul_assoc*.
rewrite mul_assoc_opp with (x := (s + T1)) (y := (s + T1)) . rewrite *mul_x_x*.
rewrite mul_assoc_opp with (x := (s + T1)) (y := s). rewrite *mul_comm* with (x := (s + T1)) (y := s).
rewrite *distr*. rewrite *mul_x_x*. rewrite mul_id_sym. rewrite *sum_x_x*. rewrite *mul_T0_x*.
repeat rewrite *mul_assoc*. rewrite *mul_comm* with (x := apply_subst $t2$ $sig2$).
repeat rewrite *mul_assoc*. rewrite mul_assoc_opp with (x := s) (y := (s + T1)).
rewrite *distr*. rewrite *mul_x_x*. rewrite mul_id_sym. rewrite *sum_x_x*. rewrite *mul_T0_x*.
repeat rewrite *sum_assoc*. rewrite sum_assoc_opp with (x := T0) (y := T0). rewrite *sum_x_x*. rewrite *sum_id*.
repeat rewrite *mul_assoc*. rewrite *mul_comm* with (x := apply_subst $t2$ $sig2$) (y := s $\times$ apply_subst $t1$ $sig2$).
repeat rewrite *mul_assoc*. rewrite mul_assoc_opp with (x := s). rewrite *mul_x_x*. reflexivity.
+ pose helper_2b as $H2$. specialize ($H2$ $t1$ $t2$ t'). apply $H2$. apply H.
+ pose helper_2a as $H2$. specialize ($H2$ $t1$ $t2$ t'). apply $H2$. apply H.
Qed.

Lemma specific_sigmas_unify:
$\forall$ (t : **term**) (tau : subst),
(unifier t tau) $\rightarrow$
(apply_subst t (build_on_list_of_vars (term_unique_vars t) t (build_id_subst (term_unique_vars t)) tau)
) == T0 .
Proof.
intros.
rewrite subs_distr_vars_ver2.
- rewrite id_subst. rewrite *mul_comm* with (x := t + T1). rewrite *distr*. rewrite *mul_x_x*. rewrite mul_id_sym. rewrite *sum_x_x*.
rewrite *sum_id*.
unfold unifier in H. rewrite H. rewrite mul_T0_x_sym. reflexivity.

```
  - apply sub_term_id.
Qed.

Lemma lownheim_unifies:
  ∀ (t : term) (tau : subst),
  (unifier t tau) →
  (apply_subst t (build_lowenheim_subst t tau)) == T0.
Proof.
intros. unfold build_lowenheim_subst. apply specific_sigmas_unify. apply H.
Qed.
```

3.3 Proof that Lownheim's algorithm produces a most general unifier

3.3.a Proof that Lownheim's algorithm produces a reproductive unifier

```
Lemma lowenheim_rephrase1_easy :
  ∀ (l : list var) (x : var) (sig1 : subst) (sig2 : subst) (s : term),
  (In x l) →
  (apply_subst (VAR x) (build_on_list_of_vars l s sig1 sig2)) ==
  (s + T1) × (apply_subst (VAR x) sig1 ) + s × (apply_subst (VAR x) sig2 ).
Proof.
intros.
induction l.
- simpl. unfold In in H. destruct H.
- apply elt_in_list in H. destruct H.
  + simpl. destruct (beq_nat a x) as []eqn:?.
    { rewrite H. reflexivity. }
    { pose proof beq_nat_false as H2. specialize (H2 a x).
      specialize (H2 Heqb). intuition. symmetry in H. specialize (H2 H). inversion
H2. }
  + simpl. destruct (beq_nat a x) as []eqn:?.
    { symmetry in Heqb. pose proof beq_nat_eq as H2. specialize (H2 a x). specialize
(H2 Heqb). rewrite H2.
      reflexivity. }
    { apply IHl. apply H. }
Qed.

Lemma helper_3a:
∀ (x: var) (l: list var),
In x l →
  apply_subst (VAR x) (build_id_subst l) == VAR x.
Proof.
intros. induction l.
 - unfold build_id_subst. simpl. reflexivity.
 - apply elt_in_list in H. destruct H.
   + simpl. destruct (beq_nat a x) as []eqn:?.
```

{ rewrite H. reflexivity. }
{ pose *proof* beq_nat_false as $H2$. specialize ($H2$ a x).
specialize ($H2$ $Heqb$). intuition. symmetry in H. specialize ($H2$ H). inversion $H2$. }
+ simpl. destruct (beq_nat a x) as []eqn:?.
{ symmetry in $Heqb$. pose *proof* beq_nat_eq as $H2$. specialize ($H2$ a x). specialize ($H2$ $Heqb$). rewrite $H2$.
reflexivity. }
{ apply IHl. apply H. }
Qed.

Lemma lowenheim_rephrase1 :
∀ (t : **term**) (tau : subst) (x : var),
(unifier t tau) →
(In x (term_unique_vars t)) →
(apply_subst (VAR x) (build_lowenheim_subst t tau)) ==
(t + T1) × (VAR x) + t × (apply_subst (VAR x) tau).
Proof.
intros.
unfold build_lowenheim_subst. pose *proof* lowenheim_rephrase1_easy as $H1$.
specialize ($H1$ (term_unique_vars t) x (build_id_subst (term_unique_vars t)) tau t).
rewrite helper_3a in $H1$.
- apply $H1$. apply $H0$.
- apply $H0$.
Qed.

Lemma lowenheim_rephrase2_easy :
∀ (l : **list** var) (x : var) ($sig1$: subst) ($sig2$: subst) (s : **term**),
¬ (In x l) →
(apply_subst (VAR x) (build_on_list_of_vars l s $sig1$ $sig2$)) ==
(VAR x).
Proof.
intros. unfold not in H.
induction l.
- simpl. reflexivity.
- simpl. pose *proof* elt_not_in_list as $H2$. specialize ($H2$ x a l). unfold not in $H2$.
specialize ($H2$ H). destruct $H2$.
destruct (beq_nat a x) as []eqn:?.
+ symmetry in $Heqb$. apply beq_nat_eq in $Heqb$. symmetry in $Heqb$. specialize ($H0$ $Heqb$). destruct $H0$.
+ simpl in IHl. apply IHl. apply $H1$.
Qed.

Lemma lowenheim_rephrase2 :

```
  ∀ (t : term) (tau : subst) (x : var),
  (unifier t tau) →
  ¬ (In x (term_unique_vars t)) →
  (apply_subst (VAR x) (build_lowenheim_subst t tau)) ==
  (VAR x).
Proof.
intros. unfold build_lowenheim_subst. pose proof lowenheim_rephrase2_easy as H2.
specialize (H2 (term_unique_vars t) x (build_id_subst (term_unique_vars t)) tau t).
specialize (H2 H0). apply H2.
Qed.

Lemma lowenheim_reproductive:
  ∀ (t : term) (tau : subst),
  (unifier t tau) →
  reproductive_unifier t (build_lowenheim_subst t tau) .
Proof.
 intros. unfold reproductive_unifier. intros.
  pose proof var_in_out_list. specialize (H2 x (term_unique_vars t)). destruct H2.
  {
  rewrite lowenheim_rephrase1.
  - rewrite subst_sum_distr_opp. rewrite subst_mul_distr_opp. rewrite subst_mul_distr_opp.
    unfold unifier in H1. rewrite H1. rewrite mul_T0_x. rewrite subst_sum_distr_opp.
    rewrite H1. rewrite ground_term_cannot_subst.
    + rewrite sum_id. rewrite mul_id. rewrite sum_comm. rewrite sum_id. reflexivity.
    + unfold ground_term. intuition.
  - apply H.
  - apply H2.
  }
  { rewrite lowenheim_rephrase2.
    - reflexivity.
    - apply H.
    - apply H2.
  }
Qed.
```

3.3.b lowenheim builder gives a most general unifier

```
Lemma lowenheim_most_general_unifier:
  ∀ (t : term) (tau : subst),
  (unifier t tau) →
  most_general_unifier t (build_lowenheim_subst t tau) .
Proof.
intros. apply reproductive_is_mgu. apply lowenheim_reproductive. apply H.
Qed.
```

3.4 extension to include Main function and subst_option

```
Definition subst_option_is_some (so : subst_option) : bool :=
  match so with
  | Some_subst s ⇒ true
  | None_subst ⇒ false
  end.

Definition convert_to_subst (so : subst_option) : subst :=
  match so with
  | Some_subst s ⇒ s
  | None_subst ⇒ nil
  end.

Lemma find_unifier_is_unifier:
 ∀ (t : term),
  (unifiable t) → (unifier t (convert_to_subst (find_unifier t))).
Proof.
intros. induction t.
{
  simpl. unfold unifier. simpl. reflexivity.
}
{
  simpl. inversion H. apply H0.
}
{
  inversion H.
Admitted.

Lemma builder_to_main:
 ∀ (t : term),
(unifiable t) → most_general_unifier t (build_lowenheim_subst t (convert_to_subst (find_unifier
t))) →
 most_general_unifier t (convert_to_subst (Lowenheim_Main t)) .
Proof.
Admitted.

Lemma lowenheim_main_most_general_unifier:
 ∀ (t: term),
 (unifiable t) → most_general_unifier t (convert_to_subst (Lowenheim_Main t)).
Proof.
 intros. apply builder_to_main.
 - apply H.
 - apply lowenheim_most_general_unifier. apply find_unifier_is_unifier. apply H.
Qed.
```

Chapter 5

Library B_Unification.poly

```
Require Import Arith.
Require Import List.
Import ListNotations.
Require Import FunctionalExtensionality.
Require Import Sorting.
Require Import Permutation.
Import Nat.
```

```
Require Export terms.
```

5.1 Introduction

Another way of representing the terms of a unification problem is as polynomials and monomials. A monomial is a set of variables multiplied together, and a polynomial is a set of monomials added together. By following the ten axioms set forth in B-unification, we can transform any term to this form.

Since one of the rules is x * x = x, we can guarantee that there are no repeated variables in any given monomial. Similarly, because x + x = 0, we can guarantee that there are no repeated monomials in a polynomial. Because of these properties, as well as the commutativity of addition and multiplication, we can represent both monomials and polynomials as unordered sets of variables and monomials, respectively. This file serves to implement such a representation.

5.2 Monomials and Polynomials

5.2.1 Data Type Definitions

A monomial is simply a list of variables, with variables as defined in terms.v.

```
Definition mono := list var.
```

```
Definition mono_eq_dec := (list_eq_dec Nat.eq_dec).
```

A polynomial, then, is a list of monomials.

```
Definition poly := list mono.
```

5.2.2 Comparisons of monomials and polynomials

For the sake of simplicity when comparing monomials and polynomials, we have opted for a solution that maintains the lists as sorted. This allows us to simultaneously ensure that there are no duplicates, as well as easily comparing the sets with the standard Coq equals operator over lists.

Ensuring that a list of nats is sorted is easy enough. In order to compare lists of sorted lists, we'll need the help of another function:

```
Fixpoint lex {T : Type} (cmp : T → T → comparison) (l1 l2 : list T)
                 : comparison :=
  match l1, l2 with
  | [], [] ⇒ Eq
  | [], _ ⇒ Lt
  | _, [] ⇒ Gt
  | h1 :: t1, h2 :: t2 ⇒
      match cmp h1 h2 with
      | Eq ⇒ lex cmp t1 t2
      | c ⇒ c
      end
  end.
```

There are some important but relatively straightforward properties of this function that are useful to prove. First, reflexivity:

```
Theorem lex_nat_refl : ∀ (l : list nat), lex compare l l = Eq.
Proof.
  intros.
  induction l.
  - simpl. reflexivity.
  - simpl. rewrite compare_refl. apply IHl.
Qed.
```

Next, antisymmetry. This allows us to take a predicate or hypothesis about the comparison of two polynomials and reverse it. For example, a $<$ b implies b $>$ a.

```
Theorem lex_nat_antisym : ∀ (l1 l2 : list nat),
  lex compare l1 l2 = CompOpp (lex compare l2 l1).
Proof.
  intros l1.
  induction l1.
  - intros. simpl. destruct l2; reflexivity.
```

```
  - intros. simpl. destruct l2.
    + simpl. reflexivity.
    + simpl. destruct (a ?= n) eqn:H;
      rewrite compare_antisym in H;
      rewrite CompOpp_iff in H; simpl in H;
      rewrite H; simpl.
      × apply IHl1.
      × reflexivity.
      × reflexivity.
Qed.

Lemma lex_eq : ∀ n m,
  lex compare n m = Eq ↔ n = m.
Proof.
  intros n. induction n; induction m; intros.
  - split; reflexivity.
  - split; intros; inversion H.
  - split; intros; inversion H.
  - split; intros; simpl in H.
    + destruct (a ?= a0) eqn:Hcomp; try inversion H. f_equal.
      × apply compare_eq_iff in Hcomp; auto.
      × apply IHn. auto.
    + inversion H. simpl. rewrite compare_refl.
      rewrite ← H2. apply IHn. reflexivity.
Qed.

Lemma lex_neq : ∀ n m,
  lex compare n m = Lt ∨ lex compare n m = Gt ↔ n ≠ m.
Proof.
  intros n. induction n; induction m.
  - simpl. split; intro. inversion H; inversion H0. contradiction.
  - simpl. split; intro. intro. inversion H0. auto.
  - simpl. split; intro. intro. inversion H0. auto.
  - clear IHm. split; intros.
    + destruct H; intro; apply lex_eq in H0; rewrite H in H0; inversion H0.
    + destruct (a ?= a0) eqn:Hcomp.
      × simpl. rewrite Hcomp. apply IHn. apply compare_eq_iff in Hcomp.
        rewrite Hcomp in H. intro. apply H. rewrite H0. reflexivity.
      × left. simpl. rewrite Hcomp. reflexivity.
      × right. simpl. rewrite Hcomp. reflexivity.
Qed.

Lemma lex_neq' : ∀ n m,
  (lex compare n m = Lt → n ≠ m) ∧
  (lex compare n m = Gt → n ≠ m).
```

```
Proof.
  intros n m. split.
  - intros. apply lex_neq. auto.
  - intros. apply lex_neq. auto.
Qed.

Lemma lex_rev_eq : ∀ n m,
  lex compare n m = Eq ↔ lex compare m n = Eq.
Proof.
  intros n m. split; intro; rewrite lex_nat_antisym in H; unfold CompOpp in H.
  - destruct (lex compare m n) eqn:H0; inversion H. reflexivity.
  - destruct (lex compare n m) eqn:H0; inversion H. reflexivity.
Qed.

Lemma lex_rev_lt_gt : ∀ n m,
  lex compare n m = Lt ↔ lex compare m n = Gt.
Proof.
  intros n m. split; intro; rewrite lex_nat_antisym in H; unfold CompOpp in H.
  - destruct (lex compare m n) eqn:H0; inversion H. reflexivity.
  - destruct (lex compare n m) eqn:H0; inversion H. reflexivity.
Qed.
```

Lastly is a property over lists. The comparison of two lists stays the same if the same new element is added onto the front of each list. Similarly, if the item at the front of two lists is equal, removing it from both does not chance the lists' comparison.

```
Theorem lex_nat_cons : ∀ (l1 l2 : list nat) n,
  lex compare l1 l2 = lex compare (n::l1) (n::l2).
Proof.
  intros. simpl. rewrite compare_refl. reflexivity.
Qed.

Hint Resolve lex_nat_refl lex_nat_antisym lex_nat_cons.
```

5.2.3 Stronger Definitions

Because as far as Coq is concerned any list of natural numbers is a monomial, it is necessary to define a few more predicates about monomials and polynomials to ensure our desired properties hold. Using these in proofs will prevent any random list from being used as a monomial or polynomial.

Monomials are simply sorted lists of natural numbers.

```
Definition is_mono (m : mono) : Prop := Sorted lt m.
```

Polynomials are sorted lists of lists, where all of the lists in the polynomail are monomials.

```
Definition is_poly (p : poly) : Prop :=
```

```
  Sorted (fun m n ⇒ lex compare m n = Lt) p ∧ ∀ m, In m p → is_mono m.
Hint Unfold is_mono is_poly.
Hint Resolve NoDup_cons NoDup_nil Sorted_cons.
Definition vars (p : poly) : list var :=
  nodup var_eq_dec (concat p).

Lemma NoDup_vars : ∀ (p : poly),
  NoDup (vars p).
Proof.
  intros p. unfold vars. apply NoDup_nodup.
Qed.
Lemma no_vars_is_ground : ∀ p,
  is_poly p →
  vars p = [] →
  p = [] ∨ p = [[]].
Proof.
Admitted.
Lemma in_mono_in_vars : ∀ x p,
  (∀ m : mono, In m p → ¬ In x m) ↔ ¬ In x (vars p).
Proof. Admitted.
```

There are a few userful things we can prove about these definitions too. First, every element in a monomial is guaranteed to be less than the elements after it.

```
Lemma mono_order : ∀ x y m,
  is_mono (x :: y :: m) →
  x < y.
Proof.
  unfold is_mono.
  intros x y m H.
  apply Sorted_inv in H as [].
  apply HdRel_inv in H0.
  apply H0.
Qed.
```

Similarly, if x :: m is a monomial, then m is also a monomial.

```
Lemma mono_cons : ∀ x m,
  is_mono (x :: m) →
  is_mono m.
Proof.
  unfold is_mono.
  intros x m H. apply Sorted_inv in H as []. apply H.
Qed.
```

The same properties hold for is_poly as well; any list in a polynomial is guaranteed to be less than the lists after it.

```
Lemma poly_order : ∀ m n p,
  is_poly (m :: n :: p) →
  lex compare m n = Lt.
Proof.
  unfold is_poly.
  intros.
  destruct H.
  apply Sorted_inv in H as [].
  apply HdRel_inv in H1.
  apply H1.
Qed.
```

And if m :: p is a polynomial, we know both that p is a polynomial and that m is a monomial.

```
Lemma poly_cons : ∀ m p,
  is_poly (m :: p) →
  is_poly p ∧ is_mono m.
Proof.
  unfold is_poly.
  intros.
  destruct H.
  apply Sorted_inv in H as [].
  split.
  - split.
    + apply H.
    + intros. apply H0, in_cons, H2.
  - apply H0, in_eq.
Qed.
```

Lastly, for completeness, nil is both a polynomial and monomial.

```
Lemma nil_is_mono :
  is_mono [].
Proof.
  unfold is_mono. auto.
Qed.

Lemma nil_is_poly :
  is_poly [].
Proof.
  unfold is_poly. split.
  - auto.
  - intro; contradiction.
```

```
Qed.

Lemma one_is_poly :
  is_poly [[]].
Proof.
  unfold is_poly. split.
  - auto.
  - intro. intro. simpl in H. destruct H.
    + rewrite ← H. apply nil_is_mono.
    + inversion H.
Qed.

Lemma var_is_poly : ∀ x,
  is_poly [[x]].
Proof.
  intros x. unfold is_poly. split.
  - apply Sorted_cons; auto.
  - intros m H. simpl in H; destruct H; inversion H.
    unfold is_mono. auto.
Qed.

Hint Resolve mono_order mono_cons poly_order poly_cons nil_is_mono nil_is_poly
  var_is_poly one_is_poly.
```

5.3 Functions over Monomials and Polynomials

```
Module Import VARSORT := NATSORT.

Fixpoint nodup_cancel {A} Aeq_dec (l : list A) : list A :=
  match l with
  | [] ⇒ []
  | x::xs ⇒
    let count := (count_occ Aeq_dec xs x) in
    let xs' := (remove Aeq_dec x (nodup_cancel Aeq_dec xs)) in
    if (even count) then x::xs' else xs'
  end.

Lemma In_remove : ∀ {A:Type} Aeq_dec a b (l:list A),
  In a (remove Aeq_dec b l) → In a l.
Proof.
  intros A Aeq_dec a b l H. induction l as [|c l IHl].
  - contradiction.
  - destruct (Aeq_dec b c) eqn:Heq; simpl in H; rewrite Heq in H.
    + right. auto.
    + destruct H; [rewrite H; intuition | right; auto].
Qed.
```

```
Lemma StronglySorted_remove : ∀ {A:Type} Aeq_dec Rel a (l:list A),
  StronglySorted Rel l → StronglySorted Rel (remove Aeq_dec a l).
Proof.
Admitted.

Lemma not_In_remove : ∀ (A:Type) Aeq_dec a (l : list A),
  ¬ In a l → (remove Aeq_dec a l) = l.
Proof.
Admitted.

Lemma remove_Sorted_eq : ∀ (A:Type) Aeq_dec x Rel (l l':list A),
  NoDup l →
  NoDup l' →
  Sorted Rel l →
  Sorted Rel l' →
  remove Aeq_dec x l = remove Aeq_dec x l' →
  l = l'.
Proof.
Admitted.

Lemma remove_distr_app : ∀ (A:Type) Aeq_dec x (l l':list A),
  remove Aeq_dec x (l ++ l') = remove Aeq_dec x l ++ remove Aeq_dec x l'.
Proof.
Admitted.

Lemma nodup_cancel_in : ∀ (A:Type) Aeq_dec a (l:list A),
  In a (nodup_cancel Aeq_dec l) → In a l.
Proof.
  intros A Aeq_dec a l H. induction l as [|b l IHl].
  - contradiction.
  - simpl in H. destruct (Aeq_dec a b).
    + rewrite e. intuition.
    + right. apply IHl. destruct (even (count_occ Aeq_dec l b)).
      × simpl in H. destruct H. rewrite H in n. contradiction.
        apply In_remove in H. auto.
      × apply In_remove in H. auto.
Qed.

Lemma NoDup_remove : ∀ (A:Type) Aeq_dec a (l:list A),
  NoDup l → NoDup (remove Aeq_dec a l).
Proof.
  intros A Aeq_dec a l H. induction l.
  - simpl. auto.
  - simpl. destruct (Aeq_dec a a0).
    + apply IHl. apply NoDup_cons_iff in H. intuition.
    + apply NoDup_cons.
```

```
      × apply NoDup_cons_iff in H as []. intro. apply H.
        apply (In_remove Aeq_dec a0 a l H1).
      × apply IHl. apply NoDup_cons_iff in H; intuition.
Qed.
Lemma NoDup_nodup_cancel : ∀ (A:Type) Aeq_dec (l:list A),
NoDup (nodup_cancel Aeq_dec l).
Proof.
  induction l as [|a l' Hrec]; simpl.
  - constructor.
  - destruct (even (count_occ Aeq_dec l' a)); simpl.
    + apply NoDup_cons; [apply remove_In | apply NoDup_remove; auto].
    + apply NoDup_remove; auto.
Qed.
Lemma Sorted_nodup_cancel : ∀ (A:Type) Aeq_dec Rel (l:list A),
  Relations_1.Transitive Rel →
  Sorted Rel l →
  Sorted Rel (nodup_cancel Aeq_dec l).
Proof.
  intros A Aeq_dec Rel l Ht H. apply Sorted_StronglySorted in H; auto.
  apply StronglySorted_Sorted. induction l.
  - auto.
  - simpl. apply StronglySorted_inv in H as []. destruct (even (count_occ Aeq_dec l a)).
    + apply SSorted_cons.
      × apply StronglySorted_remove. apply IHl. apply H.
      × admit.
    + apply StronglySorted_remove. apply IHl. apply H.
Admitted.
Lemma no_nodup_NoDup : ∀ (A:Type) Aeq_dec (l:list A),
  NoDup l →
  nodup Aeq_dec l = l.
Proof.
Admitted.
Lemma no_nodup_cancel_NoDup : ∀ (A:Type) Aeq_dec (l:list A),
  NoDup l →
  nodup_cancel Aeq_dec l = l.
Proof.
Admitted.
Lemma count_occ_Permutation : ∀ (A:Type) Aeq_dec a (l l':list A),
  Permutation l l' →
  count_occ Aeq_dec l a = count_occ Aeq_dec l' a.
Proof.
```

```
Admitted.
Lemma Permutation_not_In : ∀ (A:Type) a (l l':list A),
  Permutation l l' →
  ¬ In a l →
  ¬ In a l'.
Proof.
Admitted.
Lemma nodup_cancel_Permutation : ∀ (A:Type) Aeq_dec (l l':list A),
  Permutation l l' →
  Permutation (nodup_cancel Aeq_dec l) (nodup_cancel Aeq_dec l').
Proof.
  intros A Aeq_dec l. Admitted.
Require Import Orders.
Module MonoOrder <: TotalLeBool.
  Definition t := mono.
  Definition leb x y :=
    match lex compare x y with
    | Lt ⇒ true
    | Eq ⇒ true
    | Gt ⇒ false
    end.
  Infix "<=m" := leb (at level 35).
  Theorem leb_total : ∀ a1 a2, (a1 ≤m a2 = true) ∨ (a2 ≤m a1 = true).
  Proof.
    intros n m. unfold "<=m". destruct (lex compare n m) eqn:Hcomp; auto.
    apply lex_rev_lt_gt in Hcomp. rewrite Hcomp. auto.
  Qed.
End MonoOrder.
Module Import MonoSort := Sort MonoOrder.
Lemma VarOrder_Transitive :
  Relations_1.Transitive (fun x y : nat ⇒ is_true (NatOrder.leb x y)).
Proof.
Admitted.
Lemma MonoOrder_Transitive :
  Relations_1.Transitive (fun x y : list nat ⇒ is_true (MonoOrder.leb x y)).
Proof.
  unfold Relations_1.Transitive, is_true, MonoOrder.leb.
  induction x, y, z; intros; try reflexivity; simpl in *.
  - inversion H.
  - inversion H.
  - inversion H0.
```

```
- destruct (a ?= n) eqn:Han.
  + apply compare_eq_iff in Han. rewrite Han. destruct (n ?= n0) eqn:Hn0.
    × apply (IHx _ _ H H0).
    × reflexivity.
    × inversion H0.
  + destruct (n ?= n0) eqn:Hn0.
    × apply compare_eq_iff in Hn0. rewrite ← Hn0. rewrite Han. reflexivity.
    × apply compare_lt_iff in Han. apply compare_lt_iff in Hn0.
      apply (lt_trans a n n0 Han) in Hn0. apply compare_lt_iff in Hn0.
      rewrite Hn0. reflexivity.
    × inversion H0.
  + inversion H.
Qed.

Lemma NoDup_neq : ∀ {X:Type} (m : list X) a b,
  NoDup (a :: b :: m) →
  a ≠ b.
Proof.
  intros X m a b Hdup. apply NoDup_cons_iff in Hdup as [].
  apply NoDup_cons_iff in H0 as []. intro. apply H. simpl. auto.
Qed.

Lemma HdRel_le_lt : ∀ a m,
  HdRel (fun n m ⇒ is_true (leb n m)) a m ∧ NoDup (a::m) → HdRel lt a m.
Proof.
  intros a m []. remember (fun n m ⇒ is_true (leb n m)) as le.
  destruct m.
  - apply HdRel_nil.
  - apply HdRel_cons. apply HdRel_inv in H.
    apply (NoDup_neq _ a n) in H0; intuition. rewrite Heqle in H.
    unfold is_true in H. apply leb_le in H. destruct (a ?= n) eqn:Hcomp.
    + apply compare_eq_iff in Hcomp. contradiction.
    + apply compare_lt_iff in Hcomp. apply Hcomp.
    + apply compare_gt_iff in Hcomp. apply leb_correct_conv in Hcomp.
      apply leb_correct in H. rewrite H in Hcomp. inversion Hcomp.
Qed.

Lemma VarSort_Sorted : ∀ (m : mono),
  Sorted (fun n m ⇒ is_true (leb n m)) m ∧ NoDup m → Sorted lt m.
Proof.
  intros m []. remember (fun n m ⇒ is_true (leb n m)) as le.
  induction m.
  - apply Sorted_nil.
  - apply Sorted_inv in H. apply Sorted_cons.
    + apply IHm.
```

```
      × apply H.
      × apply NoDup_cons_iff in H0. apply H0.
    + apply HdRel_le_lt. split.
      × rewrite ← Heqle. apply H.
      × apply H0.
Qed.
Lemma Sorted_VarSorted : ∀ (m : mono),
  Sorted lt m →
  Sorted (fun n m ⇒ is_true (leb n m)) m.
Proof.
  intros m H. induction H.
  - apply Sorted_nil.
  - apply Sorted_cons.
    + apply IHSorted.
    + destruct l.
      × apply HdRel_nil.
      × apply HdRel_cons. apply HdRel_inv in H0. apply lt_le_incl in H0.
        apply leb_le in H0. apply H0.
Qed.
Lemma In_sorted : ∀ a l,
  In a l ↔ In a (sort l).
Proof.
  intros a l. pose (MonoSort.Permuted_sort l). split; intros Hin.
  - apply (Permutation_in _ p Hin).
  - apply (Permutation_in' (Logic.eq_refl a) p). auto.
Qed.
Lemma HdRel_mono_le_lt : ∀ a p,
  HdRel (fun n m ⇒ is_true (MonoOrder.leb n m)) a p ∧ NoDup (a::p) →
  HdRel (fun n m ⇒ lex compare n m = Lt) a p.
Proof.
  intros a p []. remember (fun n m ⇒ is_true (MonoOrder.leb n m)) as le.
  destruct p.
  - apply HdRel_nil.
  - apply HdRel_cons. apply HdRel_inv in H.
    apply (NoDup_neq _ a l) in H0; intuition. rewrite Heqle in H.
    unfold is_true in H. unfold MonoOrder.leb in H.
    destruct (lex compare a l) eqn:Hcomp.
    + apply lex_eq in Hcomp. contradiction.
    + reflexivity.
    + inversion H.
Qed.
```

```
Lemma MonoSort_Sorted : ∀ (p : poly),
  Sorted (fun n m ⇒ is_true (MonoOrder.leb n m)) p ∧ NoDup p →
  Sorted (fun n m ⇒ lex compare n m = Lt) p.
Proof.
  intros p []. remember (fun n m ⇒ is_true (MonoOrder.leb n m)) as le.
  induction p.
  - apply Sorted_nil.
  - apply Sorted_inv in H. apply Sorted_cons.
    + apply IHp.
      × apply H.
      × apply NoDup_cons_iff in H0. apply H0.
    + apply HdRel_mono_le_lt. split.
      × rewrite ← Heqle. apply H.
      × apply H0.
Qed.

Lemma Sorted_MonoSorted : ∀ (p : poly),
  Sorted (fun n m ⇒ lex compare n m = Lt) p →
  Sorted (fun n m ⇒ is_true (MonoOrder.leb n m)) p.
Proof.
  intros p H. induction H.
  - apply Sorted_nil.
  - apply Sorted_cons.
    + apply IHSorted.
    + destruct l.
      × apply HdRel_nil.
      × apply HdRel_cons. apply HdRel_inv in H0. unfold MonoOrder.leb.
        rewrite H0. auto.
Qed.

Lemma NoDup_MonoSorted : ∀ (p : poly),
  Sorted (fun n m ⇒ lex compare n m = Lt) p →
  NoDup p.
Proof.
Admitted.

Lemma NoDup_VarSorted : ∀ (m : mono),
  Sorted lt m → NoDup m.
Proof.
Admitted.

Lemma NoDup_VarSort : ∀ (m : mono),
  NoDup m → NoDup (VarSort.sort m).
Proof.
  intros m Hdup. pose (VarSort.Permuted_sort m).
```

```
  apply (Permutation_NoDup p Hdup).
Qed.

Lemma NoDup_MonoSort : ∀ (p : poly),
  NoDup p → NoDup (MonoSort.sort p).
Proof.
  intros p Hdup. pose (MonoSort.Permuted_sort p).
  apply (Permutation_NoDup p0 Hdup).
Qed.

Definition make_mono (l : list nat) : mono :=
  VarSort.sort (nodup var_eq_dec l).

Definition make_poly (l : list mono) : poly :=
  MonoSort.sort (nodup_cancel mono_eq_dec (map make_mono l)).

Lemma make_mono_is_mono : ∀ m,
  is_mono (make_mono m).
Proof.
  intros m. unfold is_mono, make_mono. apply VarSort_Sorted. split.
  + apply VarSort.LocallySorted_sort.
  + apply NoDup_VarSort. apply NoDup_nodup.
Qed.

Lemma make_poly_is_poly : ∀ p,
  is_poly (make_poly p).
Proof.
  intros p. unfold is_poly, make_poly. split.
  - apply MonoSort_Sorted. split.
    + apply MonoSort.LocallySorted_sort.
    + apply NoDup_MonoSort. apply NoDup_nodup_cancel.
  - intros m Hm. apply In_sorted in Hm. apply nodup_cancel_in in Hm.
    apply in_map_iff in Hm. destruct Hm. destruct H. rewrite ← H.
    apply make_mono_is_mono.
Qed.

Lemma make_mono_In : ∀ x m,
  In x (make_mono m) → In x m.
Proof.
  intros x m H. unfold make_mono in H. pose (VarSort.Permuted_sort (nodup var_eq_dec
m)).
  apply Permutation_sym in p. apply (Permutation_in _ p) in H. apply nodup_In in H.
auto.
Qed.

Lemma no_make_mono : ∀ m,
  is_mono m →
  make_mono m = m.
```

```
Proof.
Admitted.

Lemma remove_is_mono : ∀ x m,
  is_mono m →
  is_mono (remove var_eq_dec x m).
Proof.
Admitted.

Lemma no_map_make_mono : ∀ p,
  (∀ m, In m p → is_mono m) →
  map make_mono p = p.
Proof.
Admitted.

Lemma unsorted_poly : ∀ p,
  NoDup p →
  (∀ m, In m p → is_mono m) →
  nodup_cancel mono_eq_dec (map make_mono p) = p.
Proof.
  intros p Hdup Hin. rewrite no_map_make_mono; auto.
  apply no_nodup_cancel_NoDup; auto.
Qed.

Definition addPP (p q : poly) : poly :=
  make_poly (p ++ q).

Definition distribute {A} (l m : list (list A)) : list (list A) :=
  concat (map (fun a:(list A) ⇒ (map (app a) l)) m).

Definition mulPP (p q : poly) : poly :=
  make_poly (distribute p q).

Lemma addPP_is_poly : ∀ p q,
  is_poly (addPP p q).
Proof.
  intros p q. apply make_poly_is_poly.
Qed.

Lemma leb_both_eq : ∀ x y,
  is_true (MonoOrder.leb x y) →
  is_true (MonoOrder.leb y x) →
  x = y.
Proof.
  intros x y H H0. unfold is_true, MonoOrder.leb in *.
  destruct (lex compare y x) eqn:Hyx; destruct (lex compare x y) eqn:Hxy;
  try (apply lex_rev_lt_gt in Hxy; rewrite Hxy in Hyx; inversion Hyx);
  try (apply lex_rev_lt_gt in Hyx; rewrite Hxy in Hyx; inversion Hyx);
```

```
  try inversion H; try inversion H0.
  apply lex_eq in Hxy; auto.
Qed.

Lemma Permutation_incl : ∀ {A} (l m : list A),
  Permutation l m → incl l m ∧ incl m l.
Proof.
  intros A l m H. apply Permutation_sym in H as H0. split.
  + unfold incl. intros a. apply (Permutation_in _ H).
  + unfold incl. intros a. apply (Permutation_in _ H0).
Qed.

Lemma incl_cons_inv : ∀ (A:Type) (a:A) (l m : list A),
  incl (a :: l) m → In a m ∧ incl l m.
Proof.
  intros A a l m H. split.
  - unfold incl in H. apply H. intuition.
  - unfold incl in *. intros b Hin. apply H. intuition.
Qed.

Lemma Forall_In : ∀ (A:Type) (l:list A) a Rel,
  In a l → Forall Rel l → Rel a.
Proof.
  intros A l a Rel Hin Hfor. apply (Forall_forall Rel l); auto.
Qed.

Lemma Permutation_Sorted_mono_eq : ∀ (m n : mono),
  Permutation m n →
  Sorted (fun n m ⇒ is_true (leb n m)) m →
  Sorted (fun n m ⇒ is_true (leb n m)) n →
  m = n.
Proof.
  intros m n Hp Hsl Hsm. generalize dependent n.
  induction m; induction n; intros.
  - reflexivity.
  - apply Permutation_nil in Hp. auto.
  - apply Permutation_sym, Permutation_nil in Hp. auto.
  - clear IHn. apply Permutation_incl in Hp as Hp'. destruct Hp'.
    destruct (a ?= a0) eqn:Hcomp.
    + apply compare_eq_iff in Hcomp. rewrite Hcomp in *.
      apply Permutation_cons_inv in Hp. f_equal; auto.
      apply IHm.
      × apply Sorted_inv in Hsl. apply Hsl.
      × apply Hp.
      × apply Sorted_inv in Hsm. apply Hsm.
```

```
    + apply compare_lt_iff in Hcomp as Hneq. apply incl_cons_inv in H. destruct H.
      apply Sorted_StronglySorted in Hsm. apply StronglySorted_inv in Hsm as [].
      × simpl in H. destruct H; try (rewrite H in Hneq; apply lt_irrefl in Hneq;
contradiction).
        pose (Forall_In _ _ _ _ H H3). simpl in i. unfold is_true in i.
        apply leb_le in i. apply lt_not_le in Hneq. contradiction.
      × apply VarOrder_Transitive.
    + apply compare_gt_iff in Hcomp as Hneq. apply incl_cons_inv in H0. destruct H0.
      apply Sorted_StronglySorted in Hsl. apply StronglySorted_inv in Hsl as [].
      × simpl in H0. destruct H0; try (rewrite H0 in Hneq; apply gt_irrefl in Hneq;
contradiction).
        pose (Forall_In _ _ _ _ H0 H3). simpl in i. unfold is_true in i.
        apply leb_le in i. apply lt_not_le in Hneq. contradiction.
      × apply VarOrder_Transitive.
Qed.

Lemma Permutation_sort_mono_eq : ∀ (l m:mono),
  Permutation l m ↔ VarSort.sort l = VarSort.sort m.
Proof.
  intros l m. split; intros H.
  - assert (H0 : Permutation (VarSort.sort l) (VarSort.sort m)).
    + apply Permutation_trans with (l:=(VarSort.sort l)) (l':=m) (l'':=(VarSort.sort m)).
      × apply Permutation_sym. apply Permutation_sym in H.
        apply (Permutation_trans H (VarSort.Permuted_sort l)).
      × apply VarSort.Permuted_sort.
    + apply (Permutation_Sorted_mono_eq _ _ H0 (VarSort.LocallySorted_sort l) (VarSort.LocallySorted_sort
m)).
  - assert (Permutation (VarSort.sort l) (VarSort.sort m)).
    + rewrite H. apply Permutation_refl.
    + pose (VarSort.Permuted_sort l). pose (VarSort.Permuted_sort m).
      apply (Permutation_trans p) in H0. apply Permutation_sym in p0.
      apply (Permutation_trans H0) in p0. apply p0.
Qed.

Lemma Permutation_Sorted_eq : ∀ (l m : list mono),
  Permutation l m →
  Sorted (fun x y ⇒ is_true (MonoOrder.leb x y)) l →
  Sorted (fun x y ⇒ is_true (MonoOrder.leb x y)) m →
  l = m.
Proof.
  intros l m Hp Hsl Hsm. generalize dependent m.
  induction l; induction m; intros.
  - reflexivity.
  - apply Permutation_nil in Hp. auto.
```

```
  - apply Permutation_sym, Permutation_nil in Hp. auto.
  - clear IHm. apply Permutation_incl in Hp as Hp'. destruct Hp'.
    destruct (lex compare a a0) eqn:Hcomp.
    + apply lex_eq in Hcomp. rewrite Hcomp in *.
      apply Permutation_cons_inv in Hp. f_equal; auto.
      apply IHl.
      × apply Sorted_inv in Hsl. apply Hsl.
      × apply Hp.
      × apply Sorted_inv in Hsm. apply Hsm.
    + apply lex_neq' in Hcomp as Hneq. apply incl_cons_inv in H. destruct H.
      apply Sorted_StronglySorted in Hsm. apply StronglySorted_inv in Hsm as [].
      × simpl in H. destruct H; try (rewrite H in Hneq; contradiction).
        pose (Forall_In _ _ _ _ H H3). simpl in i. unfold is_true in i.
        unfold MonoOrder.leb in i. apply lex_rev_lt_gt in Hcomp.
        rewrite Hcomp in i. inversion i.
      × apply MonoOrder_Transitive.
    + apply lex_neq' in Hcomp as Hneq. apply incl_cons_inv in H0. destruct H0.
      apply Sorted_StronglySorted in Hsl. apply StronglySorted_inv in Hsl as [].
      × simpl in H0. destruct H0; try (rewrite H0 in Hneq; contradiction).
        pose (Forall_In _ _ _ _ H0 H3). simpl in i. unfold is_true in i.
        unfold MonoOrder.leb in i. rewrite Hcomp in i. inversion i.
      × apply MonoOrder_Transitive.
Qed.

Lemma Permutation_sort_eq : ∀ l m,
  Permutation l m ↔ sort l = sort m.
Proof.
  intros l m. split; intros H.
  - assert (H0 : Permutation (sort l) (sort m)).
    + apply Permutation_trans with (l:=(sort l)) (l':=m) (l'':=(sort m)).
      × apply Permutation_sym. apply Permutation_sym in H.
        apply (Permutation_trans H (Permuted_sort l)).
      × apply Permuted_sort.
    + apply (Permutation_Sorted_eq _ _ H0 (LocallySorted_sort l) (LocallySorted_sort m)).
  - assert (Permutation (sort l) (sort m)).
    + rewrite H. apply Permutation_refl.
    + pose (Permuted_sort l). pose (Permuted_sort m).
      apply (Permutation_trans p) in H0. apply Permutation_sym in p0.
      apply (Permutation_trans H0) in p0. apply p0.
Qed.

Lemma sort_app_comm : ∀ l m,
  sort (l ++ m) = sort (m ++ l).
Proof.
```

```
  intros l m. pose (Permutation.Permutation_app_comm l m).
  apply Permutation_sort_eq. auto.
Qed.

Lemma sort_nodup_cancel_assoc : ∀ l,
  sort (nodup_cancel mono_eq_dec l) = nodup_cancel mono_eq_dec (sort l).
Proof.
  intros l. apply Permutation_Sorted_eq.
  - pose (Permuted_sort (nodup_cancel mono_eq_dec l)). apply Permutation_sym in p.
    apply (Permutation_trans p). clear p. apply NoDup_Permutation.
    + apply NoDup_nodup_cancel.
    + apply NoDup_nodup_cancel.
    + intros x. split.
      * intros H. apply Permutation_in with (l:=(nodup_cancel mono_eq_dec l)).
        apply nodup_cancel_Permutation. apply Permuted_sort. auto.
      * intros H. apply Permutation_in with (l:=(nodup_cancel mono_eq_dec (sort l))).
        apply nodup_cancel_Permutation. apply Permutation_sym. apply Permuted_sort.
auto.
  - apply LocallySorted_sort.
  - apply Sorted_nodup_cancel.
    + apply MonoOrder_Transitive.
    + apply LocallySorted_sort.
Qed.

Lemma addPP_comm : ∀ p q,
  addPP p q = addPP q p.
Proof.
  intros p q. unfold addPP, make_poly. repeat rewrite map_app.
  repeat rewrite sort_nodup_cancel_assoc. rewrite sort_app_comm.
  reflexivity.
Qed.

Hint Unfold addPP mulPP.

Lemma mulPP_l_r : ∀ p q r,
  p = q →
  mulPP p r = mulPP q r.
Proof.
  intros p q r H. rewrite H. reflexivity.
Qed.

Lemma mulPP_0 : ∀ p,
  mulPP [] p = [].
Proof.
  intros p. unfold mulPP, distribute. simpl.
Admitted.
```

```
Lemma addPP_0 : ∀ p,
  is_poly p →
  addPP [] p = p.
Proof.
  intros p Hpoly. unfold addPP. simpl.
Admitted.

Lemma addPP_0r : ∀ p,
  is_poly p →
  addPP p [] = p.
Proof.
  intros p Hpoly. unfold addPP. simpl.
Admitted.

Lemma addPP_p_p : ∀ p,
  addPP p p = [].
Proof.
Admitted.

Lemma addPP_assoc : ∀ p q r,
  addPP (addPP p q) r = addPP p (addPP q r).
Proof.
Admitted.

Lemma mulPP_1r : ∀ p,
  is_poly p →
  mulPP p [[]] = p.
Proof.
Admitted.

Lemma mulPP_assoc : ∀ p q r,
  mulPP (mulPP p q) r = mulPP p (mulPP q r).
Proof.
Admitted.

Lemma mulPP_comm : ∀ p q,
  mulPP p q = mulPP q p.
Proof.
Admitted.

Lemma mulPP_p_p : ∀ p,
  mulPP p p = p.
Proof.
Admitted.

Lemma mulPP_distr_addPP : ∀ p q r,
  mulPP (addPP p q) r = addPP (mulPP p r) (mulPP q r).
Proof.
```

```
Admitted.

Lemma mulPP_distr_addPPr : ∀ p q r,
  mulPP r (addPP p q) = addPP (mulPP r p) (mulPP r q).
Proof.
Admitted.

Lemma mulPP_is_poly : ∀ p q,
  is_poly (mulPP p q).
Proof. Admitted.

Lemma mulPP_mono_cons : ∀ x m,
  is_mono (x :: m) →
  mulPP [[x]] [m] = [x :: m].
Proof.
Admitted.

Lemma addPP_poly_cons : ∀ m p,
  is_poly (m :: p) →
  addPP [m] p = m :: p.
Proof.
Admitted.

Hint Resolve addPP_is_poly mulPP_is_poly.


Lemma mulPP_addPP_1 : ∀ p q r,
  mulPP (addPP (mulPP p q) r) (addPP [[]] q) =
  mulPP (addPP [[]] q) r.
Proof.
  intros p q r. unfold mulPP.
Admitted.

Lemma partition_filter_fst {X} p l :
  fst (partition p l) = @filter X p l.
Proof.
  induction l; simpl.
  - trivial.
  - rewrite ← IHl.
    destruct (partition p l); simpl.
    destruct (p a); now simpl.
Qed.

Lemma partition_filter_fst' : ∀ {X} p (l t f : list X),
    partition p l = (t, f) →
    t = @filter X p l .
Proof.
  intros X p l t f H.
```

```
  rewrite ← partition_filter_fst.
  now rewrite H.
Qed.
Definition neg {X:Type} := fun (f:X→bool) ⇒ fun (a:X) ⇒ (negb (f a)).
Lemma neg_true_false : ∀ {X} (p:X→bool) (a:X),
  (p a) = true ↔ neg p a = false.
Proof.
  intros X p a. unfold neg. split; intro.
  - rewrite H. auto.
  - destruct (p a); intuition.
Qed.
Lemma neg_false_true : ∀ {X} (p:X→bool) (a:X),
  (p a) = false ↔ neg p a = true.
Proof.
  intros X p a. unfold neg. split; intro.
  - rewrite H. auto.
  - destruct (p a); intuition.
Qed.
Lemma partition_filter_snd {X} p l :
  snd (partition p l) = @filter X (neg p) l.
Proof.
  induction l; simpl.
  - reflexivity.
  - rewrite ← IHl.
    destruct (partition p l); simpl.
    destruct (p a) eqn:Hp.
    + simpl. apply neg_true_false in Hp. rewrite Hp; auto.
    + simpl. apply neg_false_true in Hp. rewrite Hp; auto.
Qed.
Lemma partition_filter_snd' : ∀ {X} p (l t f : list X),
  partition p l = (t, f) →
  f = @filter X (neg p) l.
Proof.
  intros X p l t f H.
  rewrite ← partition_filter_snd.
  now rewrite H.
Qed.
Lemma incl_vars_addPP : ∀ xs p q,
  incl (vars p) xs ∧ incl (vars q) xs ↔ incl (vars (addPP p q)) xs.
Proof. Admitted.
Lemma incl_vars_mulPP : ∀ xs p q,
```

```
  incl (vars p) xs ∧ incl (vars q) xs ↔ incl (vars (mulPP p q)) xs.
Proof. Admitted.

Lemma incl_nil : ∀ {X:Type} (l:list X),
  incl l [] ↔ l = [].
Proof. Admitted.

Lemma part_add_eq : ∀ f p l r,
  is_poly p →
  partition f p = (l, r) →
  p = addPP l r.
Proof.
  intros f p l r Hpoly Hpart. induction l.
  - rewrite addPP_0. unfold partition in Hpart. simpl.
Admitted.

Lemma part_fst_true : ∀ X p (l t f : list X),
  partition p l = (t, f) →
  (∀ a, In a t → p a = true).
Proof.
  intros X p l t f Hpart a Hin.
  assert (Hf: t = filter p l).
  - now apply partition_filter_fst' with f.
  - assert (Hass := filter_In p a l).
    apply Hass.
    now rewrite ← Hf.
Qed.

Lemma part_snd_false : ∀ X p (x t f : list X),
  partition p x = (t, f) →
  (∀ a, In a f → p a = false).
Proof.
  intros X p l t f Hpart a Hin.
  assert (Hf: f = filter (neg p) l).
  - now apply partition_filter_snd' with t.
  - assert (Hass := filter_In (neg p) a l).
    rewrite ← neg_false_true in Hass.
    apply Hass.
    now rewrite ← Hf.
Qed.

Lemma part_Sorted : ∀ {X:Type} (c:X→X→Prop) f p,
  Sorted c p →
  ∀ l r, partition f p = (l, r) →
  Sorted c l ∧ Sorted c r.
Proof.
```

```
  intros X c f p Hsort. induction p.
  - simpl.
Admitted.

Lemma part_is_poly : ∀ f p l r,
  is_poly p →
  partition f p = (l, r) →
  is_poly l ∧ is_poly r.
Proof.
  intros f p l r Hpoly Hpart. destruct Hpoly. split; split.
  - apply (part_Sorted _ _ _ H _ _ Hpart).
  - intros m Hin. apply H0. apply elements_in_partition with (x:=m) in Hpart.
    apply Hpart; auto.
  - apply (part_Sorted _ _ _ H _ _ Hpart).
  - intros m Hin. apply H0. apply elements_in_partition with (x:=m) in Hpart.
    apply Hpart; auto.
Qed.

Lemma addPP_cons : ∀ (m:mono) (p:poly),
  HdRel (fun m n ⇒ lex compare m n = Lt) m p →
  addPP [m] p = m :: p.
Proof. Admitted.
```

Chapter 6

Library B_Unification.poly_unif

```
Require Import ListSet.
Require Import List.
Import ListNotations.
Require Import Arith.

Require Export poly.

Definition repl := (prod var poly).

Definition subst := list repl.

Definition inDom (x : var) (s : subst) : bool :=
  existsb (beq_nat x) (map fst s).

Fixpoint appSubst (s : subst) (x : var) : poly :=
  match s with
  | [] ⇒ [[x]]
   | (y, p) :: s' ⇒ if (x =? y) then p else (appSubst s' x)
  end.

Fixpoint substM (s : subst) (m : mono) : poly :=
  match m with
  | [] ⇒ [[]]
   | x :: m ⇒ mulPP (appSubst s x) (substM s m)
  end.

Fixpoint substP (s : subst) (p : poly) : poly :=
  match p with
  | [] ⇒ []
  | m :: p' ⇒ addPP (substM s m) (substP s p')
  end.

Lemma substP_distr_mulPP : ∀ p q s,
  substP s (mulPP p q) = mulPP (substP s p) (substP s q).
Proof.
```

```
Admitted.
Lemma substP_distr_addPP : ∀ p q s,
  substP s (addPP p q) = addPP (substP s p) (substP s q).
Proof.
Admitted.
Lemma substM_cons : ∀ x m,
  ¬ In x m →
  ∀ q s, substM ((x, q) :: s) m = substM s m.
Proof.
  intros. induction m.
  - auto.
  - simpl. f_equal.
    + destruct (a =? x) eqn:H0.
      × symmetry in H0. apply beq_nat_eq in H0. exfalso. simpl in H.
        apply H. left. auto.
      × auto.
    + apply IHm. intro. apply H. right. auto.
Qed.
Lemma substP_cons : ∀ x p,
  (∀ m, In m p → ¬ In x m) →
  ∀ q s, substP ((x, q) :: s) p = substP s p.
Proof.
  intros. induction p.
  - auto.
  - simpl. f_equal.
    + apply substM_cons. apply H. left. auto.
    + apply IHp. intros. apply H. right. auto.
Qed.
Lemma substP_1 : ∀ s,
  substP s [[]] = [[]].
Proof.
  intros. simpl. rewrite addPP_0r; auto.
Qed.
Lemma substP_is_poly : ∀ s p,
  is_poly (substP s p).
Proof.
  intros. unfold substP. destruct p; auto.
Qed.
Hint Resolve substP_is_poly.
Definition unifier (s : subst) (p : poly) : Prop :=
  substP s p = [].
```

```
Definition unifiable (p : poly) : Prop :=
  ∃ s, unifier s p.
Definition subst_comp (s t u : subst) : Prop :=
  ∀ x,
  substP t (substP s [[x]]) = substP u [[x]].
Definition more_general (s t : subst) : Prop :=
  ∃ u, subst_comp s u t.
Definition mgu (s : subst) (p : poly) : Prop :=
  unifier s p ∧
  ∀ t,
  unifier t p →
  more_general s t.
Definition reprod_unif (s : subst) (p : poly) : Prop :=
  unifier s p ∧
  ∀ t,
  unifier t p →
  subst_comp s t t.
Lemma subst_comp_poly : ∀ s t u,
  (∀ x, substP t (substP s [[x]]) = substP u [[x]]) →
  ∀ p,
  is_poly p →
  substP t (substP s p) = substP u p.
Proof.
Admitted.
Lemma reprod_is_mgu : ∀ p s,
  reprod_unif s p →
  mgu s p.
Proof.
  unfold mgu, reprod_unif, more_general, subst_comp.
  intros p s [].
  split; auto.
  intros.
  ∃ t.
  intros.
  apply H0.
  auto.
Qed.
Lemma empty_substM : ∀ (m : mono),
  is_mono m →
  substM [] m = [m].
Proof.
```

```
  intros. induction m.
  - auto.
  - simpl. apply mono_cons in H as H0.
    rewrite IHm; auto.
    apply mulPP_mono_cons; auto.
Qed.

Lemma empty_substP : ∀ (p : poly),
  is_poly p →
  substP [] p = p.
Proof.
  intros.
  induction p.
  - auto.
  - simpl. apply poly_cons in H as H0. destruct H0.
    rewrite IHp; auto.
    rewrite empty_substM; auto.
    apply addPP_poly_cons; auto.
Qed.

Lemma empty_unifier : unifier [] [].
Proof.
        unfold unifier. apply empty_substP.
  unfold is_poly.
  split.
  + apply Sorted.Sorted_nil.
  + intros. inversion H.
Qed.

Lemma empty_mgu : mgu [] [].
Proof.
  unfold mgu, more_general, subst_comp.
  split.
  - apply empty_unifier.
  - intros.
    ∃ t.
    intros.
    rewrite empty_substP; auto.
Qed.

Lemma empty_reprod_unif : reprod_unif [] [].
Proof.
  unfold reprod_unif, more_general, subst_comp.
  split.
  - apply empty_unifier.
```

```
  - intros.
    rewrite empty_substP; auto.
Qed.
```

Chapter 7

Library B_Unification.sve

7.1 Intro

Here we implement the algorithm for successive variable elimination. The basic idea is to remove a variable from the problem, solve that simpler problem, and build a solution from the simpler solution. The algorithm is recursive, so variables are removed and problems generated until we are left with either of two problems; 1 =B 0 or 0 =B 0. In the former case, the whole original problem is not unifiable. In the latter case, the problem is solved without any need to substitute since there are no variables. From here, we begin the process of building up substitutions until we reach the original problem.

7.2 Eliminating Variables

This section deals with the problem of removing a variable x from a term t. The first thing to notice is that t can be written in polynomial form p. This polynomial is just a set of monomials, and each monomial a set of variables. We can now seperate the polynomials into two sets qx and r. The term qx will be the set of monomials in p that contain the variable x. The term q, or the quotient, is qx with the x removed from each monomial. The term r, or the remainder, will be the monomials that do not contain x. The original term can then be written as $x \times q + r$.

Implementing this procedure is pretty straightforward. We define a function div_by_var that produces two polynomials given a polynomial p and a variable x to eliminate from it. The first step is dividing p into qx and r which is performed using a partition over p with the predicate has_var. The second step is to remove x from qx using the helper elim_var which just maps over the given polynomial removing the given variable.

```
Definition has_var (x : var) := existsb (beq_nat x).

Definition elim_var (x : var) (p : poly) : poly :=
  make_poly (map (remove var_eq_dec x) p).

Definition div_by_var (x : var) (p : poly) : prod poly poly :=
```

```
  let (qx, r) := partition (has_var x) p in
  (elim_var x qx , r).
```

We would also like to prove some lemmas about varaible elimination that will be helpful in proving the full algorithm correct later. The main lemma below is div_eq, which just asserts that after eliminating x from p into q and r the term can be put back together as in $p = x \times q + r$. This fact turns out to be rather hard to prove and needs the help of 10 or so other sudsidiary lemmas.

```
Lemma elim_var_not_in_rem : ∀ x p r,
  elim_var x p = r →
  (∀ m, In m r → ¬ In x m).
Proof.
  intros.
  unfold elim_var in H.
  unfold make_poly in H.
  rewrite ← H in H0.
  apply In_sorted in H0.
  apply nodup_cancel_in in H0.
  rewrite map_map in H0.
  apply in_map_iff in H0 as [n []].
  rewrite ← H0.
  intro.
  apply make_mono_In in H2.
  apply remove_In in H2.
  auto.
Qed.

Lemma elim_var_poly : ∀ x p,
  is_poly (elim_var x p).
Proof.
  intros.
  unfold elim_var.
  apply make_poly_is_poly.
Qed.

Lemma NoDup_map_remove : ∀ x p,
  is_poly p →
  (∀ m, In m p → In x m) →
  NoDup (map (remove var_eq_dec x) p).
Proof.
  intros x p Hp Hx. induction p.
  - simpl. auto.
  - simpl. apply NoDup_cons.
    + intro. apply in_map_iff in H. destruct H as [y []]. assert (y = a).
```

```
      × apply poly_cons in Hp. destruct Hp. unfold is_poly in H1. destruct H1.
        apply H3 in H0. apply (remove_Sorted_eq _ var_eq_dec x lt); auto.
        – apply NoDup_VarSorted in H0. auto.
        – apply NoDup_VarSorted in H2. auto.
      × rewrite H1 in H0. unfold is_poly in Hp. destruct Hp.
        apply NoDup_MonoSorted in H2 as H4. apply NoDup_cons_iff in H4 as [].
        contradiction.
    + apply IHp.
      × apply poly_cons in Hp. apply Hp.
      × intros m H. apply Hx. intuition.
Qed.

Lemma elim_var_map_remove_Permutation : ∀ p x,
  is_poly p →
  (∀ m, In m p → In x m) →
  Permutation (elim_var x p)
              (map (remove var_eq_dec x) p).
Proof.
  intros p x H H0. destruct p as [|a p].
  - simpl. unfold elim_var, make_poly, MonoSort.sort. auto.
  - simpl. unfold elim_var. simpl. unfold make_poly. pose (MonoSort.Permuted_sort
(nodup_cancel mono_eq_dec (map make_mono (remove var_eq_dec x a :: map (remove
var_eq_dec x) p)))).
    assert (Permutation (nodup_cancel mono_eq_dec (map make_mono (remove var_eq_dec
x a :: map (remove var_eq_dec x) p))) (remove var_eq_dec x a :: map (remove var_eq_dec
x) p)).
    + clear p0. rewrite unsorted_poly.
      × apply Permutation_refl.
      × rewrite ← map_cons. apply NoDup_map_remove; auto.
      × apply poly_cons in H. intros m Hin. destruct Hin.
        – rewrite ← H1. apply remove_is_mono. apply H.
        – apply in_map_iff in H1 as [y []]. rewrite ← H1. apply remove_is_mono.
          destruct H. unfold is_poly in H. destruct H. apply H4. auto.
    + apply Permutation_sym in p0. apply (Permutation_trans p0 H1).
Qed.

Lemma concat_map : ∀ {A B:Type} (f:A→B) (l:list A),
  concat (map (fun a ⇒ [f a]) l) = map f l.
Proof.
  intros A B f l. induction l.
  - auto.
  - simpl. f_equal. apply IHl.
Qed.

Lemma NoDup_map_app : ∀ x l,
```

```
  is_poly l →
  (∀ m, In m l → ¬ In x m) →
  NoDup (map make_mono (map (fun a : list var ⇒ a ++ [x]) l)).
Proof.
  intros x l Hp Hin. induction l.
  - simpl. auto.
  - simpl. apply NoDup_cons.
    + intros H. rewrite map_map in H. apply in_map_iff in H as [m []]. assert (a=m).
      × apply poly_cons in Hp as []. apply Permutation_Sorted_mono_eq.
        – apply Permutation_sort_mono_eq in H. rewrite no_nodup_NoDup in H.
          rewrite no_nodup_NoDup in H.
          ++ pose (Permutation_cons_append m x). pose (Permutation_cons_append a
x).
             apply (Permutation_trans p) in H. apply Permutation_sym in p0.
             apply (Permutation_trans H) in p0. apply Permutation_cons_inv in p0.
             apply Permutation_sym. auto.
          ++ apply Permutation_NoDup with (l:=(x::a)). apply Permutation_cons_append.
             apply NoDup_cons. apply Hin. intuition. unfold is_mono in H2.
             apply NoDup_VarSorted in H2. auto.
          ++ apply Permutation_NoDup with (l:=(x::m)). apply Permutation_cons_append.
             apply NoDup_cons. apply Hin. intuition. unfold is_poly in H1.
             destruct H1. apply H3 in H0. unfold is_mono in H0.
             apply NoDup_VarSorted in H0. auto.
        – unfold is_mono in H2. apply Sorted_VarSorted. auto.
        – unfold is_poly in H1. destruct H1. apply H3 in H0. apply Sorted_VarSorted.
auto.
      × rewrite ← H1 in H0. unfold is_poly in Hp. destruct Hp.
        apply NoDup_MonoSorted in H2. apply NoDup_cons_iff in H2 as []. contradiction.
    + apply IHl. apply poly_cons in Hp. apply Hp. intros m H. apply Hin. intuition.
Qed.

Lemma mulPP_Permutation : ∀ x a0 l,
  is_poly (a0::l) →
  (∀ m, In m (a0::l) → ¬ In x m) →
  Permutation (mulPP [[x]] (a0 :: l)) ((make_mono (a0++[x]))::(mulPP [[x]] l)).
Proof.
  intros x a0 l Hp Hx. unfold mulPP, distribute. simpl. unfold make_poly.
  pose (MonoSort.Permuted_sort (nodup_cancel mono_eq_dec
        (map make_mono ((a0 ++ [x]) :: concat (map (fun a : list var ⇒ [a ++ [x]])
l))))).
  apply Permutation_sym in p. apply (Permutation_trans p). simpl map.
  rewrite no_nodup_cancel_NoDup; clear p.
  - apply perm_skip. apply Permutation_trans with (l':=(nodup_cancel mono_eq_dec (map
```

```
make_mono (concat (map (fun a : list var => [a ++ [x]]) l))))).
      + rewrite no_nodup_cancel_NoDup; auto. rewrite concat_map. apply NoDup_map_app.
        apply poly_cons in Hp. apply Hp. intros m H. apply Hx. intuition.
      + apply MonoSort.Permuted_sort.
   - rewrite <- map_cons. rewrite concat_map.
     rewrite <- map_cons with (f:=(fun a : list var => a ++ [x])).
     apply NoDup_map_app; auto.
Qed.

Lemma mulPP_map_app_permutation : forall (x:var) (l l' : poly),
   is_poly l ->
   (forall m, In m l -> ~ In x m) ->
   Permutation l l' ->
   Permutation (mulPP [[x]] l) (map (fun a => (make_mono(a ++ [x]))) l').
Proof.
   intros x l l' Hp H H0. generalize dependent l'. induction l; induction l'.
   - intros. unfold mulPP, distribute, make_poly, MonoSort.sort. simpl. auto.
   - intros. apply Permutation_nil_cons in H0. contradiction.
   - intros. apply Permutation_sym in H0. apply Permutation_nil_cons in H0. contradiction.
   - intros. clear IHl'. destruct (mono_eq_dec a a0).
     + rewrite e in *. pose (mulPP_Permutation x a0 l Hp H). apply (Permutation_trans
p). simpl.
        apply perm_skip. apply IHl.
        * clear p. apply poly_cons in Hp. apply Hp.
        * intros m Hin. apply H. intuition.
        * apply Permutation_cons_inv in H0. auto.
     + apply Permutation_incl in H0 as H1. destruct H1. apply incl_cons_inv in H1 as
[].
        destruct H1; try (rewrite H1 in n; contradiction). apply in_split in H1.
        destruct H1 as [l1 [l2]]. rewrite H1 in H0.
        pose (Permutation_middle (a0::l1) l2 a). apply Permutation_sym in p.
        simpl in p. apply (Permutation_trans H0) in p.
        apply Permutation_cons_inv in p. rewrite H1. simpl. rewrite map_app. simpl.
        pose (Permutation_middle ((make_mono (a0 ++ [x]) :: map
          (fun a1 : list var => make_mono (a1 ++ [x])) l1)) (map
          (fun a1 : list var => make_mono (a1 ++ [x])) l2) (make_mono (a++[x]))).
        simpl in p0. simpl. apply Permutation_trans with (l':=(make_mono (a ++ [x])
        :: make_mono (a0 ++ [x])
            :: map (fun a1 : list var => make_mono (a1 ++ [x])) l1 ++
               map (fun a1 : list var => make_mono (a1 ++ [x])) l2)); auto. clear p0.
        rewrite <- map_app. rewrite <- (map_cons (fun a1 : list var => make_mono (a1
++ [x])) a0 (@app (list var) l1 l2)).
        pose (mulPP_Permutation x a l Hp H). apply (Permutation_trans p0). apply perm_skip.
```

```
      apply IHl.
      × clear p0. apply poly_cons in Hp. apply Hp.
      × intros m Hin. apply H. intuition.
      × apply p.
Qed.
Lemma rebuild_map_permutation : ∀ p x,
  is_poly p →
  (∀ m, In m p → In x m) →
  Permutation (mulPP [[x]] (elim_var x p))
             (map (fun a ⇒ (make_mono(a ++ [x]))) (map (remove var_eq_dec x) p)).
Proof.
  intros p x H H0. apply mulPP_map_app_permutation.
  - apply elim_var_poly.
  - apply (elim_var_not_in_rem x p); auto.
  - apply elim_var_map_remove_Permutation; auto.
Qed.
Lemma p_map_Permutation : ∀ p x,
  is_poly p →
  (∀ m, In m p → In x m) →
  Permutation p (map (fun a ⇒ (make_mono(a ++ [x]))) (map (remove var_eq_dec x) p)).
Proof.
  intros p x H H0. rewrite map_map. induction p.
  - auto.
  - simpl. assert (make_mono (@app var (remove var_eq_dec x a) [x]) = a).
    + unfold make_mono. rewrite no_nodup_NoDup.
      × apply Permutation_Sorted_mono_eq.
        – apply Permutation_trans with (l':=(remove var_eq_dec x a ++ [x])).
          apply Permutation_sym. apply VarSort.Permuted_sort.
          pose (in_split x a). destruct e as [l1 [l2 e]]. apply H0. intuition.
          rewrite e. apply Permutation_trans with (l':=(x::remove var_eq_dec x (l1++x::l2))).
          apply Permutation_sym. apply Permutation_cons_append.
          apply Permutation_trans with (l':=(x::l1++l2)). apply perm_skip.
          rewrite remove_distr_app. replace (x::l2) with ([x]++l2); auto.
          rewrite remove_distr_app. simpl. destruct (var_eq_dec x x); try contradiction.
          rewrite app_nil_l. repeat rewrite not_In_remove; try apply Permutation_refl;
          try (apply poly_cons in H as []; unfold is_mono in H1;
          apply NoDup_VarSorted in H1; rewrite e in H1; apply NoDup_remove_2 in
H1).
          intros x2. apply H1. intuition. intros x1. apply H1. intuition.
          apply Permutation_middle.
        – apply VarSort.LocallySorted_sort.
        – apply poly_cons in H as []. unfold is_mono in H1.
```

```
          apply Sorted_VarSorted. auto.
      × apply Permutation_NoDup with (l:=(x::remove var_eq_dec x a)).
        apply Permutation_cons_append. apply NoDup_cons.
        apply remove_In. apply NoDup_remove. apply poly_cons in H as [].
        unfold is_mono in H1. apply NoDup_VarSorted. auto.
    + rewrite H1. apply perm_skip. apply IHp.
      × apply poly_cons in H. apply H.
      × intros m Hin. apply H0. intuition.
Qed.

Lemma elim_var_permutation : ∀ p x,
  is_poly p →
  (∀ m, In m p → In x m) →
  Permutation p (mulPP [[x]] (elim_var x p)).
Proof.
  intros p x H H0. pose (rebuild_map_permutation p x H H0).
  apply Permutation_sym in p0. pose (p_map_Permutation p x H H0).
  apply (Permutation_trans p1 p0).
Qed.

Lemma elim_var_mul : ∀ x p,
  is_poly p →
  (∀ m, In m p → In x m) →
  p = mulPP [[x]] (elim_var x p).
Proof.
  intros. apply Permutation_Sorted_eq.
  - apply elim_var_permutation; auto.
  - unfold is_poly in H. apply Sorted_MonoSorted. apply H.
  - pose (mulPP_is_poly [[x]] (elim_var x p)). unfold is_poly in i.
    apply Sorted_MonoSorted. apply i.
Qed.

Lemma has_var_eq_in : ∀ x m,
  has_var x m = true ↔ In x m.
Proof.
  intros.
  unfold has_var.
  rewrite existsb_exists.
  split; intros.
  - destruct H as [x0 []].
    apply Nat.eqb_eq in H0.
    rewrite H0. apply H.
  - ∃ x. rewrite Nat.eqb_eq. auto.
Qed.
```

```
Lemma part_var_eq_in : ∀ x p i o,
  partition (has_var x) p = (i, o) →
  ((∀ m, In m i → In x m) ∧
   (∀ m, In m o → ¬ In x m)).
Proof.
  intros.
  split; intros.
  - apply part_fst_true with (a:=m) in H.
    + apply has_var_eq_in. apply H.
    + apply H0.
  - apply part_snd_false with (a:=m) in H.
    + rewrite ← has_var_eq_in. rewrite H. auto.
    + apply H0.
Qed.

Lemma div_is_poly : ∀ x p q r,
  is_poly p →
  div_by_var x p = (q, r) →
  is_poly q ∧ is_poly r.
Proof.
  intros.
  unfold div_by_var in H0.
  destruct (partition (has_var x) p) eqn:Hpart.
  apply (part_is_poly _ _ _ _ H) in Hpart as Hp.
  destruct Hp as [Hpl Hpr].
  injection H0. intros Hr Hq.
  rewrite Hr in Hpr.
  apply part_var_eq_in in Hpart as [Hin Hout].
  split.
  - rewrite ← Hq. apply elim_var_poly.
  - apply Hpr.
Qed.
```

As explained earlier, given a polynomial p decomposed into a variable x, a quotient q, and a remainder r, div_eq asserts that $p = x \times q + r$.

```
Lemma div_eq : ∀ x p q r,
  is_poly p →
  div_by_var x p = (q, r) →
  p = addPP (mulPP [[x]] q) r.
Proof.
  intros x p q r HP HD.
  assert (HE := HD).
  unfold div_by_var in HE.
```

```
  destruct ((partition (has_var x) p)) as [qx r0] eqn:Hqr.
  injection HE. intros Hr Hq.
  assert (HIH: ∀ m, In m qx → In x m). intros.
  apply has_var_eq_in.
  apply (part_fst_true _ _ _ _ _ Hqr _ H).
  assert (is_poly q ∧ is_poly r) as [HPq HPr].
  apply (div_is_poly _ _ _ _ HP HD).
  assert (is_poly qx ∧ is_poly r0) as [HPqx HPr0].
  apply (part_is_poly _ _ _ _ HP Hqr).
  rewrite ← Hq.
  rewrite ← (elim_var_mul x qx HPqx HIH).
  apply (part_add_eq (has_var x) _ _ _ HP).
  rewrite ← Hr.
  apply Hqr.
Qed.

Lemma has_var_in : ∀ x m,
  In x m → has_var x m = true.
Proof.
  intros.
  unfold has_var.
  apply existsb_exists.
  ∃ x.
  split; auto.
  symmetry.
  apply beq_nat_refl.
Qed.

Lemma div_var_not_in_qr : ∀ x p q r,
  div_by_var x p = (q, r) →
  ((∀ m, In m q → ¬ In x m) ∧
   (∀ m, In m r → ¬ In x m)).
Proof.
  intros.
  unfold div_by_var in H.
  assert (∃ qxr, qxr = partition (has_var x) p) as [[qx r0] Hqxr]. eauto.
  rewrite ← Hqxr in H.
  injection H. intros Hr Hq.
  split.
  - apply (elim_var_not_in_rem _ _ _ Hq).
  - rewrite Hr in Hqxr.
    symmetry in Hqxr.
    intros. intro.
```

```
    apply has_var_in in H1.
    apply Bool.negb_false_iff in H1.
    revert H1.
    apply Bool.eq_true_false_abs.
    apply Bool.negb_true_iff.
    revert m H0.
    apply (part_snd_false _ _ _ _ _ Hqxr).
Qed.
```

The second main lemma about varaible elimination is below. Given that a term p has been decomposed into the form $x \times q + r$, we can define $p' = (q + 1) \times r$. The lemma div_build_unif states that any unifier of $p =B\ 0$ is also a unifier of $p' =B\ 0$. Much of this proof relies on the axioms of polynomial arithmetic.

This helper function build_poly is used to construct $p' = (q + 1) \times r$ given the quotient and remainder as inputs.

```
Definition build_poly (q r : poly) : poly :=
  mulPP (addPP [[]] q) r.

Lemma build_poly_is_poly : ∀ q r,
  is_poly (build_poly q r).
Proof.
  unfold build_poly. auto.
Qed.

Lemma div_build_unif : ∀ x p q r s,
  is_poly p →
  div_by_var x p = (q, r) →
  unifier s p →
  unifier s (build_poly q r).
Proof.
  unfold build_poly, unifier.
  intros x p q r s HPp HD Hsp0.
  apply (div_eq _ _ _ _ HPp) in HD as Hp.
  assert (∃ q1, q1 = addPP [[]] q) as [q1 Hq1]. eauto.
  assert (∃ sp, sp = substP s p) as [sp Hsp]. eauto.
  assert (∃ sq1, sq1 = substP s q1) as [sq1 Hsq1]. eauto.
  rewrite ← Hsp in Hsp0.
  apply (mulPP_l_r sp [] sq1) in Hsp0.
  rewrite mulPP_0 in Hsp0.
  rewrite ← Hsp0.
  rewrite Hsp, Hsq1.
  rewrite Hp, Hq1.
  rewrite ← substP_distr_mulPP.
  f_equal.
```

```
  assert (HMx: is_mono [x]). auto.
  apply (div_is_poly x p q r HPp) in HD.
  destruct HD as [HPq HPr].
  assert (is_mono [x] ∧ is_poly q). auto.
  rewrite mulPP_addPP_1.
  reflexivity.
Qed.

Lemma div_by_var_nil : ∀ x q r,
  div_by_var x [] = (q, r) →
  q = [] ∧ r = [].
Proof.
  intros x q r H. unfold div_by_var, elim_var, make_poly, MonoSort.sort in H.
  simpl in H. inversion H. auto.
Qed.

Hint Unfold vars div_by_var elim_var make_poly MonoSort.sort.
Hint Resolve div_by_var_nil.

Lemma incl_not_in : ∀ A a (l m : list A)
  (Aeq_dec : ∀ (a b : A), {a = b}+{a ≠ b}),
  incl l (a :: m) →
  ¬ In a l →
  incl l m.
Proof.
  intros A a l m Aeq_dec Hincl Hnin. unfold incl in *. intros a0 Hin.
  destruct (Aeq_dec a a0).
  - rewrite e in Hnin. contradiction.
  - simpl in Hincl. apply Hincl in Hin. destruct Hin; [contradiction | auto].
Qed.

Lemma incl_div : ∀ q r x,
  ∀ p, is_poly p →
  div_by_var x p = (q, r) →
  ∀ xs, incl (vars p) (x :: xs) →
  incl (vars q) xs ∧ incl (vars r) xs.
Proof.
  intros q r x. intros p H Hp. apply (div_eq x p q r H) in Hp as Hp'.
  intros xs Hxs. rewrite Hp' in Hxs. apply incl_vars_addPP in Hxs as [].
  apply incl_vars_mulPP in H0 as [].
  apply (incl_not_in _ _ _ _ var_eq_dec) in H2.
  apply (incl_not_in _ _ _ _ var_eq_dec) in H1.
  - split; auto.
  - apply div_var_not_in_qr in Hp as []. apply in_mono_in_vars in H4. auto.
  - apply div_var_not_in_qr in Hp as []. apply in_mono_in_vars in H3. auto.
```

```
Qed.

Lemma div_vars : ∀ x xs p q r,
  is_poly p →
  incl (vars p) (x :: xs) →
  div_by_var x p = (q, r) →
  incl (vars (build_poly q r)) xs.
Proof.
  intros x xs p q r H Hincl Hdiv. unfold build_poly.
  apply div_var_not_in_qr in Hdiv as Hin. destruct Hin as [Hinq Hinr].
  apply in_mono_in_vars in Hinq. apply in_mono_in_vars in Hinr.
  apply incl_vars_mulPP. apply (incl_div _ _ _ _ H Hdiv) in Hincl. split.
  - apply incl_vars_addPP. split.
    + unfold vars. simpl. unfold incl. intros a [].
    + apply Hincl.
  - apply Hincl.
Qed.
```

7.3 Building Substitutions

This section handles how a solution is built from subproblem solutions. Given that a term p has been decomposed into the form $x \times q + r$, we can define $p' = (q + 1) \times r$. The lemma reprod_build_subst states that if some substitution s is a reproductive unifier of $p' =B\ 0$, then we can build a substitution s' which is a reproductive unifier of $p =B\ 0$. The way s' is built from s is defined in build_subst. Another replacement is added to s of the form $x \to x \times (s(q) + 1) + s(r)$ to construct s'.

```
Definition build_subst (s : subst) (x : var) (q r : poly) : subst :=
  let q1 := addPP [[]] q in
  let q1s := substP s q1 in
  let rs := substP s r in
  let xs := (x, addPP (mulPP [[x]] q1s) rs) in
  xs :: s.

Lemma build_subst_is_unif : ∀ x p q r s,
  is_poly p →
  div_by_var x p = (q, r) →
  reprod_unif s (build_poly q r) →
  unifier (build_subst s x q r) p.
Proof.
  intros x p q r s Hpoly Hdiv Hreprod.
  unfold unifier. unfold reprod_unif in Hreprod.
  destruct Hreprod as [Hunif Hreprod].
  unfold unifier in Hunif.
```

```
unfold build_poly in Hunif.
assert (Hnqr := Hdiv).
apply div_var_not_in_qr in Hnqr.
destruct Hnqr as [Hnq Hnr].
assert (HpolyQR := Hdiv).
apply div_is_poly in HpolyQR as [HpolyQ HpolyR]; auto.
apply div_eq in Hdiv; auto.

rewrite Hdiv.
rewrite substP_distr_addPP.
rewrite substP_distr_mulPP.
unfold build_subst.
rewrite (substP_cons _ _ Hnq).
rewrite (substP_cons _ _ Hnr).

assert (Hsx: (substP
      ((x,
       addPP
          (mulPP [[x]]
  (substP s (addPP [[]] q)))
          (substP s r)) :: s)
      [[x]]) = (addPP
       (mulPP [[x]]
  (substP s (addPP [[]] q)))
       (substP s r))).
  simpl. unfold inDom. simpl.
  rewrite <- beq_nat_refl. simpl.
  rewrite addPP_0r; auto.
  rewrite mulPP_1r; auto.
rewrite Hsx.

rewrite substP_distr_addPP.
rewrite substP_1.
rewrite mulPP_distr_addPPr.
rewrite mulPP_1r; auto.
rewrite mulPP_distr_addPP.
rewrite mulPP_distr_addPP.
rewrite mulPP_assoc.
rewrite mulPP_p_p.
rewrite addPP_p_p.
rewrite addPP_0; auto.
rewrite <- substP_distr_mulPP.
rewrite <- substP_distr_addPP.
rewrite <- (mulPP_1r r) at 2; auto.
rewrite mulPP_comm.
```

```
  rewrite (mulPP_comm r [[]]).
  rewrite ← mulPP_distr_addPP.
  rewrite addPP_comm; auto.
Qed.

Lemma build_subst_is_reprod : ∀ x p q r s,
  is_poly p →
  div_by_var x p = (q, r) →
  reprod_unif s (build_poly q r) →
  inDom x s = false →
  ∀ t, unifier t p →
            subst_comp (build_subst s x q r) t t.
Proof.
  intros x p q r s HpolyP Hdiv Hreprod Hin t HunifT.
  assert (HunifT' := HunifT).
  apply (div_build_unif _ _ _ _ _ HpolyP Hdiv) in HunifT'.
  unfold reprod_unif in Hreprod.
  destruct Hreprod as [HunifS Hsub_comp].
  unfold subst_comp in *.
  intros y.
  destruct (y =? x) eqn:Hyx.
  - unfold build_subst.
    assert (H: (substP
        ((x, addPP (mulPP [[x]] (substP s (addPP [[]] q))) (substP s r)) :: s)
          [[y]]) =
        (addPP (mulPP [[x]] (substP s (addPP [[]] q))) (substP s r))).
      simpl substP. unfold inDom.
      simpl existsb. rewrite Hyx. simpl.
      rewrite mulPP_1r; auto.
      rewrite addPP_0r; auto.
    rewrite H.

    rewrite substP_distr_addPP.
    rewrite substP_distr_mulPP.
    rewrite substP_distr_addPP.
    rewrite substP_distr_addPP.
    rewrite substP_1.
    assert (Hdiv2 := Hdiv).
    apply div_eq in Hdiv; auto.
    apply div_is_poly in Hdiv2 as [HpolyQ HpolyR]; auto.
    rewrite (subst_comp_poly s t t); auto.
    rewrite (subst_comp_poly s t t); auto.
    rewrite mulPP_comm.
    rewrite mulPP_distr_addPP.
```

```
    rewrite mulPP_comm.
    rewrite mulPP_1r; auto.
    rewrite (addPP_comm (substP t [[x]]) _); auto.
    rewrite addPP_assoc.
    rewrite (addPP_comm (substP t [[x]]) _ ); auto.
    rewrite ← addPP_assoc.
    rewrite ← substP_distr_mulPP.
    rewrite ← substP_distr_addPP.
    rewrite mulPP_comm.
    rewrite ← Hdiv.
    unfold unifier in HunifT.
    rewrite HunifT.
    rewrite addPP_0; auto.
    apply beq_nat_true in Hyx.
    rewrite Hyx.
    reflexivity.
  - unfold build_subst.
    rewrite substP_cons; auto.
    intros.
    inversion H; auto.
    rewrite ← H0.
    simpl. intro.
    destruct H1; auto.
    apply Nat.eqb_eq in H1.
    rewrite Hyx in H1.
    inversion H1.
Qed.

Lemma reprod_build_subst : ∀ x p q r s,
  is_poly p →
  div_by_var x p = (q , r) →
  reprod_unif s (build_poly q r) →
  inDom x s = false →
  reprod_unif (build_subst s x q r) p.
Proof.
  intros.
  unfold reprod_unif.
  split.
  - apply build_subst_is_unif; auto.
  - apply build_subst_is_reprod; auto.
Qed.
```

7.4 Recursive Algorithm

Now we define the actual algorithm of successive variable elimination. Built using five helper functions, the definition is not too difficult to construct or understand. The general idea, as mentioned before, is to remove one variable at a time, creating simpler problems. Once the simplest problem has been reached, to which the solution is already known, every solution to each subproblem can be built from the solution to the successive subproblem. Formally, given the polynomials $p = x \times q + r$ and $p' = (q + 1) \times r$, the solution to $p =B\ 0$ is built from the solution to $p' =B\ 0$. If s solves $p' =B\ 0$, then $s' = s\ U\ (x \to x \times (s(q) + 1) + s(r))$ solves $p =B\ 0$.

The function sve is the final result, but it is sveVars which actually has all of the meat. Due to Coq's rigid type system, every recursive function must be obviously terminating. This means that one of the arguments must decrease with each nested call. It turns out that Coq's type checker is unable to deduce that continually building polynomials from the quotient and remainder of previous ones will eventually result in 0 or 1. So instead we add a fuel argument that explicitly decreases per recursive call. We use the set of variables in the polynomial for this purpose, since each subsequent call has one less variable.

```
Fixpoint sveVars (varlist : list var) (p : poly) : option subst :=
  match varlist with
  | [] ⇒
      match p with
      | [] ⇒ Some []
      | _ ⇒ None
      end
  | x :: xs ⇒
      let (q, r) := div_by_var x p in
      let p' := (build_poly q r) in
      match sveVars xs p' with
      | None ⇒ None
      | Some s ⇒ Some (build_subst s x q r)
      end
  end.
```

```
Definition sve (p : poly) : option subst := sveVars (vars p) p.
```

7.5 Correctness

Finally, we must show that this algorithm is correct. As discussed in the beginning, the correctness of a unification algorithm is proven for two cases. If the algorithm produces a solution for a problem, then the solution must be most general. If the algorithm produces no solution, then the problem must not be unifiable. These statements have been formalized in the theorem sve_correct with the help of the predicates mgu and unifiable as defined in the library *poly_unif.v*. The two cases of the proof are handled seperately by the lemmas sveVars_some and sveVars_none.

```
Lemma sve_in_vars_in_unif : ∀ xs y p,
  NoDup xs →
  incl (vars p) xs →
  is_poly p →
  ¬ In y xs →
  ∀ s, sveVars xs p = Some s →
          inDom y s = false.
Proof.
  induction xs as [|x xs].
  - intros y p Hdup H H0 H1 s H2. simpl in H2. destruct p; inversion H2. auto.
  - intros y p Hdup H H0 H1 s H2.
    assert (∃ qr, div_by_var x p = qr) as [[q r] Hqr]. eauto.
    simpl in H2.
    rewrite Hqr in H2.
    destruct (sveVars xs (build_poly q r)) eqn:Hs0; inversion H2.

    assert (Hvars: incl (vars (build_poly q r)) xs).
      apply (div_vars x xs p q r H0 H Hqr).

    assert (Hpoly: is_poly (build_poly q r)). simpl.
      apply build_poly_is_poly.

    assert (Hny: ¬ In y xs).
      simpl in H1. intro. auto.

    apply NoDup_cons_iff in Hdup as Hnin. destruct Hnin as [Hnin Hdup0].
    apply (IHxs _ _ Hdup0 Hvars Hpoly Hny) in Hs0.

    unfold inDom. unfold build_subst.
    simpl.
    apply Bool.orb_false_intro.
    + apply Nat.eqb_neq. simpl in H1. intro. auto.
    + unfold inDom in Hs0. apply Hs0.
Qed.

Lemma sveVars_some : ∀ (xs : list var) (p : poly),
  NoDup xs →
  incl (vars p) xs →
  is_poly p →
  ∀ s, sveVars xs p = Some s →
          mgu s p.
Proof.
  intros xs p Hdup H H0 s H1.
  apply reprod_is_mgu.
  revert xs p Hdup H H0 s H1.
```

```
  induction xs as [|x xs].
  - intros. simpl in H1. destruct p; inversion H1.
    apply empty_reprod_unif.
  - intros.
    assert (∃ qr, div_by_var x p = qr) as [[q r] Hqr]. eauto.
    simpl in H1.
    rewrite Hqr in H1.
    destruct (sveVars xs (build_poly q r)) eqn:Hs0; inversion H1.

    assert (Hvars: incl (vars (build_poly q r)) xs).
      apply (div_vars x xs p q r H0 H Hqr).

    assert (Hpoly: is_poly (build_poly q r)).
      apply build_poly_is_poly.

    apply NoDup_cons_iff in Hdup as Hnin. destruct Hnin as [Hnin Hdup0].

    assert (Hin: inDom x s0 = false).
      apply (sve_in_vars_in_unif _ _ _ Hdup0 Hvars Hpoly Hnin _ Hs0).

    apply (IHxs _ Hdup0 Hvars Hpoly) in Hs0.
    apply (reprod_build_subst _ _ _ _ _ H0 Hqr Hs0 Hin).
Qed.

Lemma sveVars_none : ∀ (xs : list var) (p : poly),
  NoDup xs →
  incl (vars p) xs →
  is_poly p →
  sveVars xs p = None →
  ¬ unifiable p.
Proof.

  induction xs as [|x xs].
  - intros p Hdup H H0 H1. simpl in H1. destruct p; inversion H1. intro.
    unfold unifiable in H2. destruct H2. unfold unifier in H2.
    apply incl_nil in H. apply no_vars_is_ground in H; auto.
    destruct H; inversion H.
    rewrite H4 in H2.
    rewrite H5 in H2.
    rewrite substP_1 in H2.
    inversion H2.
  - intros p Hdup H H0 H1.
    assert (∃ qr, div_by_var x p = qr) as [[q r] Hqr]. eauto.
    simpl in H1.
    rewrite Hqr in H1.
    destruct (sveVars xs (build_poly q r)) eqn:Hs0; inversion H1.

    assert (Hvars: incl (vars (build_poly q r)) xs).
```

```
      apply (div_vars x xs p q r H0 H Hqr).
    assert (Hpoly: is_poly (build_poly q r)).
      apply build_poly_is_poly.
    apply NoDup_cons_iff in Hdup as Hnin. destruct Hnin as [Hnin Hdup0].
    apply (IHxs _ Hdup0 Hvars Hpoly) in Hs0.
    unfold not, unifiable in *.
    intros.
    apply Hs0.
    destruct H2 as [s Hs].
    ∃ s.
    apply (div_build_unif _ _ _ _ _ H0 Hqr Hs).
Qed.

Hint Resolve NoDup_vars incl_refl.

Lemma sveVars_correct : ∀ (p : poly),
  is_poly p →
  match sveVars (vars p) p with
  | Some s ⇒ mgu s p
  | None ⇒ ¬ unifiable p
  end.
Proof.
  intros.
  remember (sveVars (vars p) p).
  destruct o.
  - apply (sveVars_some (vars p)); auto.
  - apply (sveVars_none (vars p)); auto.
Qed.

Theorem sve_correct : ∀ (p : poly),
  is_poly p →
  match sve p with
  | Some s ⇒ mgu s p
  | None ⇒ ¬ unifiable p
  end.
Proof.
  intros.
  apply sveVars_correct.
  auto.
Qed.
```